UniDyn--Demo-02.nb

John A. Marohn
jam99@cornell.edu
Cornell University

Abstract: Use the **UniDyn** Evolver function to calculate the evolution of the magnetization of a single spin 1/2 particle under off-resonance, variable-phase irradiation. Plot the evolving magnetization for various combinations of resonance offest and irradiation phase.

Set the path to the package

Tell *Mathematica* the path to the directory containing the package.

EDIT THE FOLLOWING PATH STRING:

```
$PackagePath =
  "/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/
    unidyn";
```

YOU SHOULD NOT NEED TO EDIT ANYTHING FROM HERE ONWARDS.

Load the package

Append the package path to the system path. Before trying to load the package, ask *Mathematica* to find it. This is a test that we directed *Mathematica* to the correct directory. The output of this command should be the full system path to the UniDyn.m file.

```
$Path = AppendTo[$Path, $PackagePath];
FindFile["UniDyn`"]
```

```
/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/UniDyn/UniDyn.m
```

Now that we are confident that the path is set correctlly, load the package. Setting the global $VerboseLoad variable to True will print out the help strings for key commands in the package.

```
$VerboseLoad = True;
Needs["UniDyn`"]
```

```
You are using the version of NCAlgebra which is found in:
```

```
  /Users/jam99/Downloads/NC
You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.
NCMultiplication.m loaded
NC1SetCommands.m loaded
NCInverses.m loaded
NCTransposes.m loaded
NCAdjoints.m loaded
NCCo.m loaded
NCRoots.m loaded
NC2SetCommands.m loaded
NCCollect.m loaded
NCSubstitute.m loaded
NCMonomial.m loaded
NCSolve.m loaded
NCTools.m loaded
NC2SimplifyRational.m loaded
NC1SimplifyRational.m loaded
NCSimplifyRational.m loaded
NCComplex.m loaded
NCMatMult.m loaded
NCDiff.m loaded
NCSchur.m loaded
NCAlias.m loaded
Grabs.m loaded
NCTaylorCoeff.m loaded
NCConvexity.m and NCGuts.m loaded
NCRealizationFunctions.m loaded
NCTeXForm.m loaded
NCTeX::Using 'open' as PDFViewer.
NCTeX.m loaded
NCMaster.m loaded
NCOutput.m loaded
```

```
------------------------------------------------------------
NCAlgebra - Version 4.0.6
Compatible with Mathematica Version 9

Authors:
  J. William Helton*
  Mauricio de Oliveira*
  Mark Stankus*
  Robert L. Miller#

* Math Dept, UCSD
# General Atomics Corp
  La  Jolla, CA 92093

Copyright:
  Helton and Miller June 1991
  Helton 2002
  All rights reserved.

The program was written by the authors and by:
  David Hurst, Daniel Lamm, Orlando Merino, Robert Obar,
  Henry Pfister, Mike Walker, John Wavrik, Lois Yu,
  J. Camino, J. Griffin, J. Ovall, T. Shaheen, John Shopple.
  The beginnings of the program come from eran@slac.
  Considerable recent help came from Igor Klep.


This program was written with support from
  AFOSR, NSF, ONR, Lab for Math and Statistics at UCSD,
  UCSD Faculty Mentor Program,
  and US Department of Education.
  Primary support in 2010 is from the
    NSF Division of Mathematical Sciences.

If you
  (1) are a user,
  (2) want to be a user,
  (3) refer to NCAlgebra in a publication, or
  (4) have had an interesting experience with NCAlgebra,
let us know by sending an e-mail message to

  ncalg@math.ucsd.edu.

We do not want to restrict access to NCAlgebra, but do
  want to keep track of how it is being used.

For NCAlgebra updates see the web page:

  www.math.ucsd.edu/~ncalg
------------------------------------------------------------
You are using the version of NCAlgebra which is found in:

  /Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m
```

CreateOperator::usage :

CreateOperator[] is used to batch-define a bunch of operators. Example: CreateOperator[{{Ix, Iy, Iz},{Sx,Sy,Sz}}] will create six operators; each of the operators in the first list is meant to commute with each of the operators in the second list.

CreateScalar::usage :

CreateScalar[list] is used to batch-define a bunch of scalars. The parameter list can be a single scalar or a list of scalars. Example: CreateScalar[{w1,w2}].

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

NCSort::usage : NCSort[list] sorts the operators in list into canonical order.

SortedMult::usage :
SortedMult[list] returns Mult[list$ordered], where list$ordered are the elements of list sorted into canonical order.

MultSort::usage :
MultSort[NonCommutativeMultiplyt[list]] returns returns NonCommutativeMultiply[list$ordered], where list$ordered are the elements of list sorted into canonical order.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

Comm::usage : Comm[a,b] calculates the commutator of two operators.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

SpinSingle$CreateOperators::usage :
SpinSingle$CreateOperators[Ix,Iy,Iz,L] creates Ix, Iy, and Iz angular momentum operators and defines their commutation relations. When the total angular momentum L = 1/2, additional rules are defined to simplify products of the angular momentum operators. When the total angular momentum L is unspecified, no such simplification rules are defined.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

OscSingle$CreateOperators::usage :
OscSingle$CreateOperators[aL,aR] creates a raising operator aR and a lowering operator aL for single harmonic oscillator and defines the operator commutation relations.

You are using the version of NCAlgebra which is found in:

/Users/jam99/Downloads/NC

You can now use "<< NCAlgebra`" to load NCAlgebra or "<< NCGB`" to load NCGB.

You have already loaded NCAlgebra.m

Evolve::usage :
Evolve[H,t,ρ] represents unitary evolution of the density operator ρ for a time t under the Hamiltonian H. This function expands according to simplification rules but leaves the evolution unevaluated.

Evolver::usage : Evolver[H,t,ρ(0)] calculates ρ(t) = Exp[−I H t] ρ(0) Exp[+I H t], assuming that H is time independent, according to the commutation rules followed by ρ(0) and H.

Function to help draw the magneziation

```
Clear[my$drawing];
SetAttributes[my$drawing, HoldAll];

my$drawing[func_[t_, a___], t$final_, N$step_] :=

 Module[{ρ$vector$data, ρ$arrows, axes$arrows, big$plot},

  (* Calculate a final time and a time step *)
  (* The final time point should not be included in the plot *)

  T$max = t$final * (N$step - 1) / N$step;
  T$step = t$final / N$step;

  (* Make a table of data of the form *)
  (* {{0.,{0.,0.,1.}},{0.5,{0.,-0.9,-0.5}} *)

  ρ$vector$data =
   Table[{N[t / T$max], N[func[t, a]]}, {t, 0, T$max, T$step}];

  (* Add arrows; the arrows grow from
   light to dark as time progresses in the plot*)

  ρ$arrows =
   Graphics3D[{GrayLevel[1.0 - N[#[[1]]]], {Arrowheads[0.015],
        Arrow[Tube[{{0, 0, 0}, #[[2]]}]]}}] & /@ ρ$vector$data;

  (* Add axes arrows.  Here we make
   the assumption that the magnetization vector *)
  (*  has a magnitude of 1. *)

  axes$arrows =
   Graphics3D[{Black, Arrow[Tube[{{0, 0, 0}, #}]]}] & /@
    {{0, 0, 1.25}, {0, 1.25, 0}, {1.25, 0, 0},
     {0, 0, -1.25}, {0, -1.25, 0}, {-1.25, 0, 0}};

  big$plot = Flatten[Append[ρ$arrows, axes$arrows]];

  (* Add axes labels. *)

  big$plot = Flatten[Append[big$plot,
     Graphics3D[Text[Style[z, Large], {0, 0, 1.35}]]]];
```

```
big$plot = Flatten[Append[big$plot,
    Graphics3D[Text[Style[y, Large], {0, 1.35, 0}]]]];
big$plot = Flatten[Append[big$plot,
    Graphics3D[Text[Style[x, Large], {1.35, 0, 0}]]]];

(* Plot all the arrows.  The neutral
 lighting helps making the rendering fast -- *)
(*  the default Mathematica camera has three-
 colored lights which makes funny reflections *)
(*  of off small objects like our arrows. *)

Show[big$plot, Boxed → False, ViewVertical -> {0, 0, 1},
 ViewPoint → {2.0, -1.0, 1.0}, Lighting → "Neutral"]

]
```

Examples of unitary evolution in a spin 1/2 system

Create a single spin

The assumptions define below are required for *Mathematica* to recognize $\sqrt{-\Delta^2 - \omega^2} = I\sqrt{\Delta^2 + \omega^2}$ inside an exponential. One of the variables has to be defined to be > 0 and not just ⩾ 0.

```
Clear[

 Δ,           (* Resonance offset frequency *)
 ω,           (* Rabi frequency of the applied irradiation *)
 ϕ,           (* Phase of the applied irradiation *)
 Ix, Iy, Iz,  (* Spin angular momentum operators *)
 ρ,           (* Spin density operator *)
 ρ$0,         (* Initial spin density operator *)
 H            (* Spin Hamiltonian *)]

CreateScalar[Δ, ω, ϕ];
SpinSingle$CreateOperators[Ix, Iy, Iz, L = 1 / 2];

$Assumptions =
  {Element[Δ, Reals], Δ ⩾ 0, Element[ω, Reals], ω > 0};
```

SpinSingle$CreateOperators::create : Creating spin operators.

SpinSingle$CreateOperators::comm : Adding spin commutations relations.

SpinSingle$CreateOperators::simplify : Angular momentum L = 1/2. Adding operator simplification rules.

Off-resonance variable-phase nutation

Irradiation Hamilonian written in the interaction representation. The intial density operator is parallel to I_z.

```
H = Δ Iz + ω (Cos[ϕ] Ix + Sin[ϕ] Iy);
ρ$0 = Iz;
```

Calculating the time-dependent denstity operator might take as long as 10 to 15 seconds to complete.

```
ρ [t_, Δ_, ω_, ϕ_] = Collect[
       (Evolver[H, t, ρ$0] // Simplify // ExpToTrig //
     FullSimplify),
       {Ix, Iy, Iz}];

ρ[t, Δ, ω, ϕ] /.
 {Δ -> Subscript[ω, 0], ω -> Subscript[ω, 1]}
```

$$\mathrm{Iz}\left(1+\frac{\left(-1+\mathrm{Cos}\left[t\sqrt{\omega_0^2+\omega_1^2}\right]\right)\omega_1^2}{\omega_0^2+\omega_1^2}\right)+$$

$$\mathrm{Iy}\left(-\frac{\left(-1+\mathrm{Cos}\left[t\sqrt{\omega_0^2+\omega_1^2}\right]\right)\mathrm{Sin}[\phi]\,\omega_0\,\omega_1}{\omega_0^2+\omega_1^2}-\right.$$

$$\left.\frac{\mathrm{Cos}[\phi]\,\mathrm{Sin}\left[t\sqrt{\omega_0^2+\omega_1^2}\right]\omega_1}{\sqrt{\omega_0^2+\omega_1^2}}\right)+$$

$$\mathrm{Ix}\left(-\frac{\mathrm{Cos}[\phi]\left(-1+\mathrm{Cos}\left[t\sqrt{\omega_0^2+\omega_1^2}\right]\right)\omega_0\,\omega_1}{\omega_0^2+\omega_1^2}+\right.$$

$$\left.\frac{\mathrm{Sin}[\phi]\,\mathrm{Sin}\left[t\sqrt{\omega_0^2+\omega_1^2}\right]\omega_1}{\sqrt{\omega_0^2+\omega_1^2}}\right)$$

Below we want a function that returns a triple of numbers descibing the magnetization vector. We turn the above expression for the density operator into a triple of numbers using the *Mathematica* function Coefficient. I tried using the NCAlgebra's NCCoefficient function but could not get it to work. The function below does what we want.

```
ρ$vector[t_, Δ_, ω_, ϕ_] =
 Simplify[Coefficient[ρ[t, Δ, ω, ϕ], #, 1] & /@ {Ix , Iy, Iz}]
```

$$\left\{-\frac{\Delta\,\omega\,\mathrm{Cos}[\phi]\left(-1+\mathrm{Cos}\left[t\sqrt{\Delta^2+\omega^2}\right]\right)}{\Delta^2+\omega^2}+\frac{\omega\,\mathrm{Sin}[\phi]\,\mathrm{Sin}\left[t\sqrt{\Delta^2+\omega^2}\right]}{\sqrt{\Delta^2+\omega^2}},\right.$$
$$-\frac{\Delta\,\omega\left(-1+\mathrm{Cos}\left[t\sqrt{\Delta^2+\omega^2}\right]\right)\mathrm{Sin}[\phi]}{\Delta^2+\omega^2}-\frac{\omega\,\mathrm{Cos}[\phi]\,\mathrm{Sin}\left[t\sqrt{\Delta^2+\omega^2}\right]}{\sqrt{\Delta^2+\omega^2}},$$
$$\left.\frac{\Delta^2+\omega^2\,\mathrm{Cos}\left[t\sqrt{\Delta^2+\omega^2}\right]}{\Delta^2+\omega^2}\right\}$$

Check limiting cases

On resonance, the effective field is in the x-y plane. The z magnetization will oscillate cosinusoidally while the magnetization in the x-y plane will oscillate sinusoidally.

```
ρ[t, 0, ω, ϕ] // PowerExpand // FullSimplify
```

$$\mathrm{Iz}\,\mathrm{Cos}[t\,\omega]+(-\mathrm{Iy}\,\mathrm{Cos}[\phi]+\mathrm{Ix}\,\mathrm{Sin}[\phi])\,\mathrm{Sin}[t\,\omega]$$

Apply a π pulse. Observe that the magnetization is indeed inverted.

```
ρ[π/ω, 0, ω, 0] // PowerExpand
```

$-\mathrm{Iz}$

Now apply a $\pi/2$ pulse. Appying an “x” pulse, one with a relative phase of ϕ=0, places the magnetization along the -y axis. A “y” pulse, one with a relative phase of $\phi = \pi/2$, places the magnetization along the +x axis.

```
ρ[π/(2 ω), 0, ω, #] & /@ {0, π/2} // PowerExpand
```

$\{-\mathrm{Iy}, \mathrm{Ix}\}$

Draw the magnetization

Set the rf phase to $\phi = 0$, set the Rabi frequency to $\omega = 1$, and set the resonance offest to $\Delta = 2$. The effective field has a magnitude of $\sqrt{2^2+1^2} = \sqrt{5}$, so we’ll watch the magnetization out to a time of $2\pi \big/ \sqrt{5}$ in order to capture the magnetization orbiting once around the effective field.

```
Show[my$drawing[ρ$vector[t, 2, 1, 0], 2π/√5, 36], ImageSize → Full]
```



Set the rf phase to $\phi = 0$, set the Rabi frequency to $\omega = 1$, and look at magnetization out to times equal to $\pi/2$. Vary the resonance offset and plot the magnezation.

```
Show[GraphicsGrid @@
  {{my$drawing[ρ$vector[t, #, 1, 0], π / 2, 18]} &
    /@ {0., 1., 2., 3.}}, ImageSize → Full]
```


z
y
x


z
y


x


z
y
x



Set ther resonance offset Δ = 0, set the Rabi frequency to ω = 1, and look at magnetization out to times equal to $\pi/2$. Vary the rf phase and plot the magnetization.

```
Show[GraphicsGrid @@
  {{my$drawing[ρ$vector[t, 0, 1, #], π/2, 18]} &
    /@ ({0., 45, 90, 180, 270} π/180)}, ImageSize → Full]
```




y
x


z
y
x


z
y
x



Clean up

```
(*
Clear[ω, Δ, ϕ, Ix, Iy, Iz, ρ, ρ$0, H]
*)
```