UniDyn--Study-02.nb

John A. Marohn
jam99@cornell.edu
Cornell University

Abstract: This demonstration notebook loads the **UniDyn** package, loads the **Inv** function, and runs the **Inv** function unit tests. Below we write a new function for carrying out unitary rotations, **Evolver2[]**, that works on a case-by-case basis. It is approximately 25x faster than **Evolver[]**, and handles more cases.

Set the path to the package

Check the Mathematica version number .

In[96]:=
```
$VersionNumber
```

Out[96]=
```
12.3
```

Tell *Mathematica* the path to the directory containing the packages.

EDIT THE FOLLOWING PATH STRINGS:

In[97]:=
```
$UniDynPath =
  "/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/
    unidyn";
```

In[98]:=
```
$UniDynStudyPath =
  "/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/
    studies";
```

YOU SHOULD NOT NEED TO EDIT ANYTHING FROM HERE ONWARDS.

Load and test the packages

Append the package path to the system path. Before trying to load the package, ask *Mathematica* to find it. This is a test that we directed *Mathematica* to the correct directory. The output of this command should be the full system path to the UniDyn.m file.

In[99]:=
```
$Path = AppendTo[$Path, $UniDynPath];
FindFile["UniDyn`"]
```

Out[100]=
```
/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/unidyn/UniDyn.m
```

Now that we are confident that the path is set correctly, load the package. Setting the global $VerboseLoad variable to True will print out the help strings for key commands in the package.

In[101]:=
```
$VerboseLoad = True;
Needs["UniDyn`"]
```

In[103]:=
```
$Path = AppendTo[$Path, $UniDynStudyPath];
FindFile["Inv`"]
FindFile["SpinBoson`"]
```

Out[104]=
```
/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/studies/Inv.m
```

Out[105]=
```
/Users/jam99/Dropbox/MarohnGroup__Software_Library/UniDyn/studies/SpinBoson.m
```

In[106]:=
```
$VerboseLoad = True;
Needs["Inv`"]
Needs["SpinBoson`"]
```

Extend the directory to the /studies directory and run all the tests found there.

In[109]:=
```
SetDirectory[$UniDynStudyPath];
```

In[110]:=
```
fn = FileNames["*-tests.m"];
test$report = TestReport /@ fn;
TableForm[Table[test$report[[k]], {k, 1, Length[test$report]}]]
```

```
SpinSingle$CreateOperators: Spin operators already exist.
SpinSingle$CreateOperators: Adding spin commutations relations.
SpinSingle$CreateOperators: Angular momentum L = 1/2. Adding operator simplification rules.
OscSingle$CreateOperators: Oscillator operators already exist.
OscSingle$CreateOperators: Adding oscillator commutations relations.
```

Out[112]//TableForm=

```
TestReportObject[ Title: Test Report: Evolve2-tests.m
                  Success rate: 100%   Tests run: 22 ]

TestReportObject[ Title: Test Report: Inv-tests.m
                  Success rate: 100%   Tests run: 22 ]

TestReportObject[ Title: Test Report: SpinBoson-tests.m
                  Success rate: 100%   Tests run: 2 ]
```

Create quantum optics operators

In[113]:=
```
Clear[Ix, Iy, Iz, I$p, I$m, aR, aL];
SpinBoson$CreateOperators[Ix, Iy, Iz, I$p, I$m, aR, aL];
```

SpinSingle$CreateOperators: Spin operators already exist.

SpinSingle$CreateOperators: Adding spin commutations relations.

SpinSingle$CreateOperators: Angular momentum L = 1/2. Adding operator simplification rules.

OscSingle$CreateOperators: Oscillator operators already exist.

OscSingle$CreateOperators: Adding oscillator commutations relations.

SpinBoson$CreateOperators: Creating operators.

SpinBoson$CreateOperators: Adding Ip and Im commutations relations.

SpinBoson$CreateOperators: Adding Ip and Im simplification rules.

SpinBoson$CreateOperators: Adding aL and aR normal ordering rule.

In[115]:=
```
Clear[Δ, ω1, g];
CreateScalar[{Δ, ω1, g}];
$Assumptions = {Δ ∈ Reals, ω1 ∈ PositiveReals, g ∈ PositiveReals};
```

Are these assumptions enough? We can see the following examples that Sqrt[] involving the real parameters will not simplify correctly if there is more than one real parameter in the Sqrt[]. To simplify the Sqrt[] with multiple real parameters inside, it is enough to assume that one of the parameters is positive.

In[118]:=
```
PowerExpand[Sqrt[-Δ^2]]
```

Out[118]= $i\,\Delta$

In[119]:=
```
PowerExpand[Sqrt[-Δ^2 - ω1^2]]
```

Out[119]= $\sqrt{-\Delta^2 - \omega 1^2}$

In[120]:=
```
PowerExpand[Sqrt[-Δ^2 - ω1^2], Assumptions → $Assumptions]
```

Out[120]= $i\,\sqrt{\Delta^2 + \omega 1^2}$

In[121]:=
```
Simplify[Sqrt[-Δ^2 - ω1^2] ]
```

Out[121]= $i\,\sqrt{\Delta^2 + \omega 1^2}$

A rotation involving spin raising and lowering operators that succeeds

We can now evolve the raising and lowering operators .

In[122]:=
```
{Evolver[λ Iz, t, I$p], Evolver[λ Iz, t, I$m]}
```
Out[122]= $\{e^{-i t \lambda}\, \text{I\$p},\ e^{i t \lambda}\, \text{I\$m}\}$

Two rotation involving mixed spin and harmonic oscillator operators that fail

Here is a unitary rotation that Evolver[] cannot resolve.

In[123]:=
```
terms = Evolver[λ Mult[Iz, aL], t, I$p, quiet → False]
```

$$\rho\ \text{matrix} = \begin{pmatrix} \text{I\$p} \\ -i\,\lambda\,\text{Mult}[\text{I\$p, aL}] \\ -\lambda^2\,\text{Mult}[\text{I\$p, aL, aL}] \\ i\,\lambda^3\,\text{Mult}[\text{I\$p, aL, aL, aL}] \\ \lambda^4\,\text{Mult}[\text{I\$p, aL, aL, aL, aL}] \end{pmatrix}$$

Evolver: Unrecognized evolution

Out[123]= $\{\text{I\$p},\ -i\,\lambda\,\text{Mult}[\text{I\$p, aL}],\ -\lambda^2\,\text{Mult}[\text{I\$p, aL, aL}],\ i\,\lambda^3\,\text{Mult}[\text{I\$p, aL, aL, aL}],\ \lambda^4\,\text{Mult}[\text{I\$p, aL, aL, aL, aL}]\}$

The ratio of the 1st to the 0th term is independent of I$p .

In[124]:=
```
Mult[Inv[terms[[1]]], terms[[2]]]
```
Out[124]= $-i\,\text{aL}\,\lambda$

The ratio of a higher term to the next lower term is likewise independent of I$p.

In[125]:=
```
{Mult[Inv[terms[[2]]], terms[[3]]], Mult[Inv[terms[[3]]], terms[[4]]]}
```
Out[125]= $\{-i\,\text{aL}\,\lambda,\ -i\,\text{aL}\,\lambda\}$

Here is another unitary rotation that Evolver[] cannot resolve.

In[126]:=
```
terms = Evolver[2 I g (Mult[Iz, aR] - Mult[Iz, aL]), t, I$p] // Simplify
```

Evolver: Unrecognized evolution

Out[126]= $\{\text{I\$p},\ -2\,g\,(\text{Mult}[\text{I\$p, aL}] - \text{Mult}[\text{I\$p, aR}]),$
$-4\,g^2\,(\text{I\$p} - \text{Mult}[\text{I\$p, aL, aL}] + 2\,\text{Mult}[\text{I\$p, aR, aL}] - \text{Mult}[\text{I\$p, aR, aR}]),$
$8\,g^3\,(3\,\text{Mult}[\text{I\$p, aL}] - 3\,\text{Mult}[\text{I\$p, aR}] - \text{Mult}[\text{I\$p, aL, aL, aL}] +$
$3\,\text{Mult}[\text{I\$p, aR, aL, aL}] - 3\,\text{Mult}[\text{I\$p, aR, aR, aL}] + \text{Mult}[\text{I\$p, aR, aR, aR}]),$
$16\,g^4\,(3\,\text{I\$p} - 6\,\text{Mult}[\text{I\$p, aL, aL}] + 12\,\text{Mult}[\text{I\$p, aR, aL}] - 6\,\text{Mult}[\text{I\$p, aR, aR}] +$
$\text{Mult}[\text{I\$p, aL, aL, aL, aL}] - 4\,\text{Mult}[\text{I\$p, aR, aL, aL, aL}] + 6\,\text{Mult}[\text{I\$p, aR, aR, aL, aL}] -$
$4\,\text{Mult}[\text{I\$p, aR, aR, aR, aL}] + \text{Mult}[\text{I\$p, aR, aR, aR, aR}])\}$

Notice that again the ratio of the 1st to the 0th term is independent of I$p.

In[127]:=
```
{Mult[Inv[terms[[1]]], terms[[2]]]}
```
Out[127]= $\{-2\,(\text{aL} - \text{aR})\,g\}$

Create a new Evolver algorithm

Pass 1

Strip down the existing algorithm, and print out some useful intermediate results.

In[128]:=
```
VisualComplexity := (Count[ToBoxes[#], Except[" " | "(" | ")", _String], Infinity] &)
Options[Evolver2] = {quiet → True};
```

In[130]:=
```
Evolver2[H$sym_, t$sym_, rho$sym$0_, opts : OptionsPattern[]] :=

 Module[{},

 Clear[rho$sym];
 rho$sym[0] = rho$sym$0;

 Do[
    rho$sym[k + 1] = (-I Comm[H$sym, rho$sym[k]] /. Mult → SortedMult)
       // FullSimplify[#, ComplexityFunction → VisualComplexity] &,
  {k, 0, 4}
 ];

 If[OptionValue[quiet] == False,
   Print["ρn= ", rho$sym[#] & /@ {0, 1, 2, 3, 4} // MatrixForm];
    Print["divisions = ", {Mult[Inv[rho$sym[0]], rho$sym[1]],
      Mult[Inv[rho$sym[0]], rho$sym[2]],
      Mult[Inv[rho$sym[1]], rho$sym[3]]}];
   Print["commutators = ", {rho$sym[1],
     Comm[rho$sym[0], rho$sym[1]],
     Comm[rho$sym[0], rho$sym[2]],
     Comm[rho$sym[1], rho$sym[3]]}]]

 ]
```

The 0th case to consider is when the Hamiltonian commutes with the initial condition. If this is the case we should just stop.

In[131]:= `terms = Evolver2[Δ I$p, t, I$p, quiet → False]`

$$\rho_n = \begin{pmatrix} \text{I\$p} \\ 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

Power: Infinite expression $\frac{1}{0}$ encountered.

Infinity: Indeterminate expression 0 ComplexInfinity encountered.

divisions = {0, 0, Indeterminate}

commutators = {0, 0, 0, 0}

The 1st case is when the first commutator commutes with the initial condition.

In[132]:= `terms = Evolver2[g Mult[Iz, aL], t, I$p, quiet → False]`

$$\rho_n = \begin{pmatrix} \text{I\$p} \\ -i\, g\, \text{Mult}[\text{I\$p}, \text{aL}] \\ -g^2\, \text{Mult}[\text{I\$p}, \text{aL}, \text{aL}] \\ i\, g^3\, \text{Mult}[\text{I\$p}, \text{aL}, \text{aL}, \text{aL}] \\ g^4\, \text{Mult}[\text{I\$p}, \text{aL}, \text{aL}, \text{aL}, \text{aL}] \end{pmatrix}$$

divisions = $\{-i\, \text{aL}\, g, -g^2\, \text{Mult}[\text{aL}, \text{aL}], -g^2\, \text{Mult}[\text{aL}, \text{aL}]\}$

commutators = $\{-i\, g\, \text{Mult}[\text{I\$p}, \text{aL}], 0, 0, 0\}$

The 2nd case is when the second commutator commutes with the initial condition.

In[133]:= `Evolver2[Δ Iz, t, Ix, quiet → False]`

$$\rho_n = \begin{pmatrix} \text{Ix} \\ \text{Iy}\, \Delta \\ -\text{Ix}\, \Delta^2 \\ -\text{Iy}\, \Delta^3 \\ \text{Ix}\, \Delta^4 \end{pmatrix}$$

divisions = $\{\Delta\, \text{Mult}[\text{Inv}[\text{Ix}], \text{Iy}], -\Delta^2, -\Delta^2\}$

commutators = $\{\text{Iy}\, \Delta, i\, \text{Iz}\, \Delta, 0, 0\}$

The 3nd case is when the third commutator commutes with the first commutator.

In[134]:= `Evolver2[Δ Iz + ω1 Ix, t, Iz, quiet → False]`

$$\rho_n = \begin{pmatrix} \text{Iz} \\ -\text{Iy}\, \omega 1 \\ \omega 1\, (\text{Ix}\, \Delta - \text{Iz}\, \omega 1) \\ \text{Iy}\, \omega 1\, (\Delta^2 + \omega 1^2) \\ \omega 1\, (-\text{Ix}\, \Delta + \text{Iz}\, \omega 1)\, (\Delta^2 + \omega 1^2) \end{pmatrix}$$

divisions = $\{-\omega 1\, \text{Mult}[\text{Inv}[\text{Iz}], \text{Iy}], \omega 1\, (-\omega 1 + \Delta\, \text{Mult}[\text{Inv}[\text{Iz}], \text{Ix}]), -\Delta^2 - \omega 1^2\}$

commutators = $\{-\text{Iy}\, \omega 1, i\, \text{Ix}\, \omega 1, i\, \text{Iy}\, \Delta\, \omega 1, 0\}$

Pass 2

Now implement cases, writing down the solution to the differential equations by hand.

```
In[135]:= Evolver2::unsolvable = "Unrecognized evolution";

Evolver2[H$sym_, t$sym_, rho$sym$0_, opts : OptionsPattern[]] :=

 Module[{rho$sym, divisions, commutators, ρ, w},

 rho$sym[0] = rho$sym$0;

 Do[
    rho$sym[k + 1] = (-I Comm[H$sym, rho$sym[k]] /. Mult → SortedMult)
       // FullSimplify[#, ComplexityFunction → VisualComplexity] &,
  {k, 0, 4}
 ];

  If[OptionValue[quiet] == False, Print["ρ(n) = ", rho$sym[#] & /@ {0, 1, 2, 3}]];

  If[rho$sym[1] == 0,
   If[OptionValue[quiet] == False, Print["Case 0"]];
   Return[rho$sym[0]]];

  If[Comm[rho$sym[1], H$sym ] == 0,
   If[OptionValue[quiet] == False, Print["Case 1L"]];
   Return[rho$sym[0] + rho$sym[1] t$sym]];

  commutators = {rho$sym[1],
    Comm[rho$sym[0], rho$sym[1]],
    Comm[rho$sym[0], rho$sym[2]],
    Comm[rho$sym[1], rho$sym[3]]};

  If[OptionValue[quiet] == False, Print["commutators = ", commutators]];

  divisions = {Mult[Inv[rho$sym[0]], rho$sym[1]],
     Mult[Inv[rho$sym[0]], rho$sym[2]],
     Mult[Inv[rho$sym[1]], rho$sym[3]]};

  If[OptionValue[quiet] == False, Print["divisions = ", divisions]];

  If[commutators⟦2⟧ == 0,
   If[OptionValue[quiet] == False, Print["Case 1E"]];
```

```
   ρ = Mult[Exp[divisions[[1]] t$sym], rho$sym[0]];
   Return[ρ]];

  If[commutators[[3]] == 0,
   If[OptionValue[quiet] == False, Print["Case 2E"]];
   w = PowerExpand[Sqrt[-divisions[[2]]]];
   If[OptionValue[quiet] == False, Print["w = ", w]];
   ρ = Mult[ Cos[w t$sym], rho$sym[0]] + Mult[ Sin[w t$sym], commutators[[1]] / w];
   Return[ρ]];

  If[commutators[[4]] == 0,
   If[OptionValue[quiet] == False, Print["Case 3E"]];
   w = PowerExpand[Sqrt[-divisions[[3]]]];
   If[OptionValue[quiet] == False, Print["w = ", w]];
   ρ = rho$sym[0] + Mult[ Sin[w t$sym], rho$sym[1] / w] +
     Mult[ 1 - Cos[w t$sym], rho$sym[2] / w^2];
   Return[ρ]];

  Message[Evolver2::unsolvable];
  Return[{rho$sym[#] & /@ {0, 1, 2, 3}, commutators, divisions}]
 ]
```

The 0a-th case is when the Hamiltonian commutes with the density operator.

In[137]:=
```
Evolver2[λ I$p, t, I$p, quiet → False]
```

$\rho^{(n)} = \{I\$p, 0, 0, 0\}$

Case 0

Out[137]= I$p

The 0b-th case is when the Hamiltonian commutes with the first commutator.

In[138]:=
```
Evolver2[I g 1/Sqrt[2] (aR - aL), 1, 1/Sqrt[2] (aL + aR), quiet → False] // Simplify
```

$\rho^{(n)} = \left\{\frac{aL + aR}{\sqrt{2}}, -g, 0, 0\right\}$

Case 1L

Out[138]= $\frac{aL + aR}{\sqrt{2}} - g$

The 1st case is when the first commutator commutes with the initial condition.

In[139]:=
```
Evolver2[g Mult[Iz, aL], t, I$p, quiet → False]
```

$\rho^{(n)} = \{I\$p, -i\, g\, \text{Mult}[I\$p, aL], -g^2\, \text{Mult}[I\$p, aL, aL], i\, g^3\, \text{Mult}[I\$p, aL, aL, aL]\}$

commutators = $\{-i\, g\, \text{Mult}[I\$p, aL], 0, 0, 0\}$

divisions = $\{-i\, aL\, g, -g^2\, \text{Mult}[aL, aL], -g^2\, \text{Mult}[aL, aL]\}$

Case 1E

Out[139]= $\text{Mult}[e^{-i\, aL\, g\, t}, I\$p]$

Compare to

In[140]:= **Evolver[g Mult[Iz, aL], t, I$p]**

Evolver: Unrecognized evolution

Out[140]= $\{I\$p, -i\, g\, \text{Mult}[I\$p, aL], -g^2\, \text{Mult}[I\$p, aL, aL], i\, g^3\, \text{Mult}[I\$p, aL, aL, aL], g^4\, \text{Mult}[I\$p, aL, aL, aL, aL]\}$

The 2nd case is when the second commutator commutes with the initial condition.

In[141]:= **Evolver2[Δ Iz, t, Ix, quiet → False]**

$\rho^{(n)} = \{Ix, Iy\,\Delta, -Ix\,\Delta^2, -Iy\,\Delta^3\}$

commutators = $\{Iy\,\Delta, i\, Iz\,\Delta, 0, 0\}$

divisions = $\{\Delta\, \text{Mult}[\text{Inv}[Ix], Iy], -\Delta^2, -\Delta^2\}$

Case 2E

w = Δ

Out[141]= $Ix\, \text{Cos}[t\,\Delta] + Iy\, \text{Sin}[t\,\Delta]$

Compare to

In[142]:= **Evolver[Δ Iz, t, Ix]**

Out[142]= $Ix\, \text{Cos}[t\,\Delta] + Iy\, \text{Sin}[t\,\Delta]$

The 3nd case is when the third commutator commutes with the first commutator.

In[143]:= **Evolver2[Δ Iz + ω1 Ix, t, Iz, quiet → False] // Collect[#, {Ix, Iy, Iz}, Simplify] &**

$\rho^{(n)} = \{Iz, -Iy\,\omega1, \omega1\,(Ix\,\Delta - Iz\,\omega1), Iy\,\omega1\,(\Delta^2 + \omega1^2)\}$

commutators = $\{-Iy\,\omega1, i\, Ix\,\omega1, i\, Iy\,\Delta\,\omega1, 0\}$

divisions = $\{-\omega1\, \text{Mult}[\text{Inv}[Iz], Iy], \omega1\,(-\omega1 + \Delta\, \text{Mult}[\text{Inv}[Iz], Ix]), -\Delta^2 - \omega1^2\}$

Case 3E

w = $\sqrt{\Delta^2 + \omega1^2}$

Out[143]= $$\frac{Iz\left(\Delta^2 + \omega1^2\, \text{Cos}\left[t\sqrt{\Delta^2+\omega1^2}\right]\right)}{\Delta^2+\omega1^2} + \frac{2\, Ix\,\Delta\,\omega1\, \text{Sin}\left[\frac{1}{2}\, t\sqrt{\Delta^2+\omega1^2}\right]^2}{\Delta^2+\omega1^2} - \frac{Iy\,\omega1\, \text{Sin}\left[t\sqrt{\Delta^2+\omega1^2}\right]}{\sqrt{\Delta^2+\omega1^2}}$$

Compare to

In[144]:=
```
Evolver[Δ Iz + ω1 Ix, t, Iz] // Collect[#, {Ix, Iy, Iz}, Simplify] &
```

Out[144]=
$$\frac{\text{Iz}\left(\Delta^2+\omega 1^2 \operatorname{Cos}\left[t\sqrt{\Delta^2+\omega 1^2}\right]\right)}{\Delta^2+\omega 1^2}+\frac{2\,\text{Ix}\,\Delta\,\omega 1 \operatorname{Sin}\left[\frac{1}{2}t\sqrt{\Delta^2+\omega 1^2}\right]^2}{\Delta^2+\omega 1^2}-\frac{\text{Iy}\,\omega 1 \operatorname{Sin}\left[t\sqrt{\Delta^2+\omega 1^2}\right]}{\sqrt{\Delta^2+\omega 1^2}}$$

Speed test

In[145]:=
```
timing1 = Timing[
   Evolver[Δ Iz + ω1 Ix, t, Iz]
  ]
```

Out[145]=
$$\left\{0.026488, \frac{\Delta\,(\text{Iz}\,\Delta+\text{Ix}\,\omega 1)+\omega 1\,(-\text{Ix}\,\Delta+\text{Iz}\,\omega 1)\operatorname{Cos}\left[t\sqrt{\Delta^2+\omega 1^2}\right]-\text{Iy}\,\omega 1\sqrt{\Delta^2+\omega 1^2}\operatorname{Sin}\left[t\sqrt{\Delta^2+\omega 1^2}\right]}{\Delta^2+\omega 1^2}\right\}$$

In[146]:=
```
timing2 = Timing[
   Evolver2[Δ Iz + ω1 Ix, t, Iz]
  ]
```

Out[146]=
$$\left\{0.00099, \text{Iz}+\frac{\omega 1\,(\text{Ix}\,\Delta-\text{Iz}\,\omega 1)}{\Delta^2+\omega 1^2}-\frac{\omega 1\,(\text{Ix}\,\Delta-\text{Iz}\,\omega 1)\operatorname{Cos}\left[t\sqrt{\Delta^2+\omega 1^2}\right]}{\Delta^2+\omega 1^2}-\frac{\text{Iy}\,\omega 1\operatorname{Sin}\left[t\sqrt{\Delta^2+\omega 1^2}\right]}{\sqrt{\Delta^2+\omega 1^2}}\right\}$$

The ratio of the execution times is

In[147]:=
```
Round[timing1[[1]] / timing2[[1]], 1]
```

Out[147]= 27

For this example of **Case 3E** evolution, **Evolver2[]** is approximately **28x faster** than **Evolver[]**.

Electron transfer test cases

This first case my old code handled .

In[148]:=
```
{ Evolver[-2 I g Mult[Iz, aR - aL], 1, aL],
  Evolver2[-2 I g Mult[Iz, aR - aL], 1, aL]}
```

Out[148]= {aL + 2 g Iz, aL + 2 g Iz}

These next two more complicated cases, **Evolver[]** failed to handle.

Evolver2[] handles these two cases of non-trivial spin-boson evolution!

In[149]:=
```
{ Evolver[I g (Mult[Iz, aR] - Mult[Iz, aL]), t, I$p],
  Evolver2[I g (Mult[Iz, aR] - Mult[Iz, aL]), t, I$p]}
```

Evolver: Unrecognized evolution

Out[149]=
```
{{I$p, -g (Mult[I$p, aL] - Mult[I$p, aR]),
  -g^2 (I$p - Mult[I$p, aL, aL] + 2 Mult[I$p, aR, aL] - Mult[I$p, aR, aR]),
  g^3 (3 Mult[I$p, aL] - 3 Mult[I$p, aR] - Mult[I$p, aL, aL, aL] +
     3 (Mult[I$p, aR, aL, aL] - Mult[I$p, aR, aR, aL]) + Mult[I$p, aR, aR, aR]),
  g^4 (3 I$p - 6 Mult[I$p, aL, aL] + 12 Mult[I$p, aR, aL] - 6 Mult[I$p, aR, aR] + Mult[
      I$p, aL, aL, aL, aL] - 4 Mult[I$p, aR, aL, aL, aL] + 6 Mult[I$p, aR, aR, aL, aL] -
     4 Mult[I$p, aR, aR, aR, aL] + Mult[I$p, aR, aR, aR, aR])}, Mult[e^(-((aL-aR) g t)), I$p]}
```

In[150]:=
```
{ Evolver[-2 I g Mult[Iz, aR - aL], t, I$p],
  Evolver2[-2 I g Mult[Iz, aR - aL], t, I$p]}
```

Evolver: Unrecognized evolution

Out[150]=
```
{{I$p, 2 g (Mult[I$p, aL] - Mult[I$p, aR]),
  -4 g^2 (I$p - Mult[I$p, aL, aL] + 2 Mult[I$p, aR, aL] - Mult[I$p, aR, aR]),
  -8 g^3 (3 Mult[I$p, aL] - 3 Mult[I$p, aR] - Mult[I$p, aL, aL, aL] +
     3 (Mult[I$p, aR, aL, aL] - Mult[I$p, aR, aR, aL]) + Mult[I$p, aR, aR, aR]),
  16 g^4 (3 I$p - 6 Mult[I$p, aL, aL] + 12 Mult[I$p, aR, aL] - 6 Mult[I$p, aR, aR] + Mult[
      I$p, aL, aL, aL, aL] - 4 Mult[I$p, aR, aL, aL, aL] + 6 Mult[I$p, aR, aR, aL, aL] -
     4 Mult[I$p, aR, aR, aR, aL] + Mult[I$p, aR, aR, aR, aR])}, Mult[e^(2 (aL-aR) g t), I$p]}
```

Create two spin operators (both I = 1/2)

In[151]:=
```
Clear[Ix, Iy, Iz, Sx, Sy, Sz];
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz, 1 / 2];
SpinSingle$CreateOperators[Sx, Sy, Sz, 1 / 2];
```

SpinSingle$CreateOperators: Spin operators already exist.

SpinSingle$CreateOperators: Adding spin commutations relations.

SpinSingle$CreateOperators: Angular momentum L = 1/2. Adding operator simplification rules.

SpinSingle$CreateOperators: Spin operators already exist.

SpinSingle$CreateOperators: Adding spin commutations relations.

SpinSingle$CreateOperators: Angular momentum L = 1/2. Adding operator simplification rules.

In[155]:=
```
Clear[Δ, J];
CreateScalar[{Δ, J}];
$Assumptions = {Δ ∈ Reals, J ∈ Reals};
```

This rotation shows up when calculating evolution of an Ix under a heteronuclear *J* coupling.

For the I = 1/2 case, both **Evolver[]** and **Evolver2[]** know what to do.

In[158]:=
```
{ Evolver[J Mult[Sz, Iz], t, Ix],
  Evolver2[J Mult[Sz, Iz], t, Ix]}
```

Out[158]= $\left\{\text{Ix}\,\text{Cos}\left[\frac{\text{J t}}{2}\right]+2\,\text{Mult}[\text{Iy},\text{Sz}]\,\text{Sin}\left[\frac{\text{J t}}{2}\right],\ \text{Ix}\,\text{Cos}\left[\frac{\text{J t}}{2}\right]+2\,\text{Mult}[\text{Iy},\text{Sz}]\,\text{Sin}\left[\frac{\text{J t}}{2}\right]\right\}$

Create two spin operators (one I = 1/2)

In[159]:=
```
Clear[Ix, Iy, Iz, Sx, Sy, Sz];
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz, 1 / 2];
SpinSingle$CreateOperators[Sx, Sy, Sz];
```

SpinSingle$CreateOperators: Spin operators already exist.

SpinSingle$CreateOperators: Adding spin commutations relations.

SpinSingle$CreateOperators: Angular momentum L = 1/2. Adding operator simplification rules.

SpinSingle$CreateOperators: Spin operators already exist.

SpinSingle$CreateOperators: Adding spin commutations relations.

SpinSingle$CreateOperators: No angular momentum L defined.

In[163]:=
```
Clear[Δ, J];
CreateScalar[{Δ, J}];
$Assumptions = {Δ ∈ Reals, J ∈ Reals};
```

For the general-I case, **Evolver[]** cannot handle the following rotation.

In[166]:=
```
Evolver[J Mult[Sz, Iz], t, Ix]
```

Evolver: Unrecognized evolution

Out[166]= $\{\text{Ix},\ \text{J}\,\text{Mult}[\text{Iy},\text{Sz}],\ -\text{J}^2\,\text{Mult}[\text{Ix},\text{Sz},\text{Sz}],\ -\text{J}^3\,\text{Mult}[\text{Iy},\text{Sz},\text{Sz},\text{Sz}],\ \text{J}^4\,\text{Mult}[\text{Ix},\text{Sz},\text{Sz},\text{Sz},\text{Sz}]\}$

Evolver2[] handles the general I-spin heteronuclear J-coupling evolution nicely.

In[167]:=
```
Evolver2[J Mult[Sz, Iz], t, Ix, quiet → False]
```

$\rho^{(n)}$ = {Ix, J Mult[Iy, Sz], - J^2 Mult[Ix, Sz, Sz], - J^3 Mult[Iy, Sz, Sz, Sz]}

commutators = {J Mult[Iy, Sz], ⅈ J Mult[Iz, Sz], 0, 0}

divisions = {J Mult[Inv[Ix], Iy, Sz], - J^2 Mult[Sz, Sz], - J^2 Mult[Sz, Sz]}

Case 2E

w = J $\sqrt{\text{Mult[Sz, Sz]}}$

Out[167]= Mult[Cos[J t $\sqrt{\text{Mult[Sz, Sz]}}$], Ix] + Mult[Sin[J t $\sqrt{\text{Mult[Sz, Sz]}}$], $\frac{\text{Mult[Iy, Sz]}}{\sqrt{\text{Mult[Sz, Sz]}}}$]

We could now think about using the Lowdin projection operator theorem to expand functions like Cos[J t $\sqrt{\text{Mult[Sz, Sz]}}$] and Sin[J t $\sqrt{\text{Mult[Sz, Sz]}}$].

Neato!

Create two spin operators (both general I) [unrecognized evolution]

In[168]:=
```
Clear[Ix, Iy, Iz, Sx, Sy, Sz];
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz];
SpinSingle$CreateOperators[Sx, Sy, Sz];
```

SpinSingle$CreateOperators : Spin operators already exist.

SpinSingle$CreateOperators : Adding spin commutations relations.

SpinSingle$CreateOperators : No angular momentum L defined.

SpinSingle$CreateOperators : Spin operators already exist.

SpinSingle$CreateOperators : Adding spin commutations relations.

SpinSingle$CreateOperators : No angular momentum L defined.

In[172]:=
```
Clear[Δ, J];
CreateScalar[{Δ, J}];
$Assumptions = {Δ ∈ Reals, J ∈ Reals};
```

For the general-I case, **Evolver2[]** cannot handle the following rotation. We can see that the commutators quickly get out of hand.

In[175]:=
```
Evolver2[J Mult[Sz, Iz, Iy], t, Ix]
```

Evolver2 : Unrecognized evolution

Out[175]= {{Ix, J (Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]),
 - J^2 (Mult[Ix, Iy, Iy, Sz, Sz] + Mult[Iy, Ix, Iy, Sz, Sz] +
 Mult[Iz, Ix, Iz, Sz, Sz] + Mult[Iz, Iz, Ix, Sz, Sz]),
 J^3 (2 Mult[Ix, Ix, Iy, Iy, Sz, Sz, Sz] + Mult[Ix, Iy, Ix, Iy, Sz, Sz, Sz] +

```
        Mult[Iy, Ix, Ix, Iy, Sz, Sz, Sz] - 2 Mult[Iy, Iy, Iy, Iy, Sz, Sz, Sz] -
        Mult[Iz, Ix, Ix, Iz, Sz, Sz, Sz] - Mult[Iz, Ix, Iz, Ix, Sz, Sz, Sz] -
        2 (Mult[Iz, Iz, Ix, Ix, Sz, Sz, Sz] - Mult[Iz, Iz, Iz, Iz, Sz, Sz, Sz]))},
  {J (Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]), J (2 ⅈ Mult[Iy, Iz, Sz] + 2 ⅈ Mult[Iz, Iy, Sz]),
   -J^2 (ⅈ Mult[Ix, Iy, Iz, Sz, Sz] + ⅈ Mult[Ix, Iz, Iy, Sz, Sz] -
        ⅈ Mult[Iy, Iz, Ix, Sz, Sz] - ⅈ Mult[Iz, Iy, Ix, Sz, Sz]),
   J (J^3 (-ⅈ Mult[Ix, Ix, Ix, Iz, Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Ix, Ix, Iz, Ix, Sz, Sz,
            Sz, Iy, Sz] - ⅈ Mult[Ix, Iy, Iz, Iy, Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Iy, Ix,
            Ix, Ix, Iz, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Ix, Ix, Iz, Ix, Sz, Sz, Sz, Sz] -
          ⅈ Mult[Iy, Ix, Iy, Iz, Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Ix, Iz, Iy,
            Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Iy, Iy, Ix, Iz, Iy, Sz, Sz, Sz, Sz] -
          ⅈ Mult[Iy, Iy, Iz, Ix, Iy, Sz, Sz, Sz, Sz] - 2 ⅈ Mult[Iy, Iz, Ix, Ix,
            Ix, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Iz, Ix, Iy, Sz, Sz, Sz, Iy, Sz] +
          2 ⅈ Mult[Iy, Iz, Ix, Iz, Iz, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Iz, Iy, Ix,
            Iy, Sz, Sz, Sz, Sz] + ⅈ Mult[Iy, Iz, Iz, Ix, Iz, Sz, Sz, Sz, Sz] +
          ⅈ Mult[Iy, Iz, Iz, Iz, Ix, Sz, Sz, Sz, Sz] - 2 ⅈ Mult[Iz, Ix, Ix, Ix,
            Sz, Sz, Sz, Iy, Sz] + 2 (-ⅈ Mult[Ix, Iz, Iy, Iy, Sz, Sz, Sz, Iy, Sz] -
             ⅈ Mult[Iy, Ix, Iz, Iy, Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Iz, Ix, Iy,
               Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Ix, Iy, Iy, Sz, Sz, Sz, Iy, Sz]) +
          2 ⅈ Mult[Iz, Ix, Iz, Iz, Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Iz, Iy, Ix, Iy,
            Sz, Sz, Sz, Iy, Sz] + ⅈ Mult[Iz, Iz, Ix, Iz, Sz, Sz, Sz, Iy, Sz] -
          2 (ⅈ Mult[Ix, Iz, Ix, Ix, Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Ix, Iz, Iz, Iz, Sz,
               Sz, Sz, Iy, Sz] + ⅈ Mult[Iy, Ix, Iz, Ix, Ix, Sz, Sz, Sz, Sz] - ⅈ Mult[Iy, Ix,
               Iz, Iz, Iz, Sz, Sz, Sz, Sz] + ⅈ Mult[Iy, Iz, Ix, Ix, Ix, Sz, Sz, Sz, Sz] -
             ⅈ Mult[Iy, Iz, Ix, Iz, Iz, Sz, Sz, Sz, Sz] - 2 ⅈ Mult[Iy, Iz, Iz, Ix,
               Iz, Sz, Sz, Sz, Sz] - 2 ⅈ Mult[Iy, Iz, Iz, Iz, Ix, Sz, Sz, Sz, Sz] +
             ⅈ Mult[Iz, Ix, Ix, Ix, Sz, Sz, Sz, Iy, Sz] - ⅈ Mult[Iz, Ix, Iz, Iz, Sz, Sz,
               Sz, Iy, Sz] - 2 ⅈ Mult[Iz, Iz, Ix, Iz, Sz, Sz, Sz, Iy, Sz] - 2 ⅈ Mult[Iz, Iz,
               Iz, Ix, Sz, Sz, Sz, Iy, Sz]) + ⅈ Mult[Iz, Iz, Iz, Ix, Sz, Sz, Sz, Iy, Sz]) -
      J^3 (-2 ⅈ Mult[Ix, Ix, Ix, Iy, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Ix, Iy, Ix, Ix,
            Sz, Sz, Sz, Iz, Sz] + ⅈ Mult[Ix, Iy, Iy, Iy, Sz, Sz, Sz, Iz, Sz] -
          ⅈ Mult[Iy, Ix, Ix, Ix, Sz, Sz, Sz, Iz, Sz] + ⅈ Mult[Iy, Ix, Iy, Iy,
            Sz, Sz, Sz, Iz, Sz] + 2 ⅈ Mult[Iy, Iy, Ix, Iy, Sz, Sz, Sz, Iz, Sz] -
          2 ⅈ Mult[Iz, Ix, Ix, Ix, Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Ix, Iy, Ix,
            Ix, Sz, Sz, Sz, Sz] + ⅈ Mult[Iz, Ix, Iy, Iy, Iy, Sz, Sz, Sz, Sz] -
          ⅈ Mult[Iz, Ix, Iy, Iz, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iz, Ix, Iz, Iy,
            Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iz, Iy, Ix, Ix, Ix, Sz, Sz, Sz, Sz] +
          2 (-ⅈ Mult[Ix, Ix, Ix, Iy, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Ix, Ix, Iy, Ix, Sz,
               Sz, Sz, Iz, Sz] + ⅈ Mult[Ix, Iy, Iy, Iy, Sz, Sz, Sz, Iz, Sz] + ⅈ Mult[Iy, Ix,
               Iy, Iy, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iz, Ix, Ix, Ix, Iy, Sz, Sz, Sz, Sz] -
             ⅈ Mult[Iz, Ix, Ix, Iy, Ix, Sz, Sz, Sz, Sz] + ⅈ Mult[Iz, Ix, Iy, Iy,
               Iy, Sz, Sz, Sz, Sz] + ⅈ Mult[Iz, Iy, Ix, Iy, Iy, Sz, Sz, Sz, Sz]) +
          ⅈ Mult[Iz, Iy, Ix, Iy, Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Iy, Ix, Iz,
```

```
              Sz, Sz, Sz, Iz, Sz] + 2 ⅈ Mult[Iz, Iy, Iy, Ix, Iy, Sz, Sz, Sz, Sz] -
            2 (-ⅈ Mult[Ix, Iy, Iy, Iy, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iy, Ix, Iy, Iy, Sz,
                 Sz, Sz, Iz, Sz] - ⅈ Mult[Iy, Iy, Ix, Iy, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iy, Iy,
                 Iy, Ix, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iz, Ix, Iy, Iy, Iy, Sz, Sz, Sz, Sz] -
               ⅈ Mult[Iz, Iy, Ix, Iy, Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Iy, Iy, Ix,
                 Iy, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Iy, Iy, Iy, Ix, Sz, Sz, Sz, Sz]) -
            ⅈ Mult[Iz, Iy, Iz, Ix, Sz, Sz, Sz, Iz, Sz] - ⅈ Mult[Iz, Iz, Ix, Iy,
              Iz, Sz, Sz, Sz, Sz] - ⅈ Mult[Iz, Iz, Ix, Iz, Iy, Sz, Sz, Sz, Sz] -
            ⅈ Mult[Iz, Iz, Iy, Ix, Iz, Sz, Sz, Sz, Sz] -
            ⅈ Mult[Iz, Iz, Iy, Iz, Ix, Sz, Sz, Sz, Sz] -
            2 (ⅈ Mult[Iz, Iz, Ix, Iy, Sz, Sz, Sz, Iz, Sz] +
               ⅈ Mult[Iz, Iz, Iy, Ix, Sz, Sz, Sz, Iz, Sz] + ⅈ Mult[Iz, Iz, Iz, Ix, Iy,
                 Sz, Sz, Sz, Sz] + ⅈ Mult[Iz, Iz, Iz, Iy, Ix, Sz, Sz, Sz, Sz])))},
 {J (Mult[Inv[Ix], Iy, Iy, Sz] - Mult[Inv[Ix], Iz, Iz, Sz]),
  -J^2
   (Mult[Iy, Iy, Sz, Sz] +
     Mult[Inv[Ix], Iy, Ix, Iy, Sz, Sz] +
     Mult[Inv[Ix], Iz, Ix, Iz, Sz, Sz] +
     Mult[Inv[Ix], Iz, Iz, Ix, Sz, Sz]),
  J^2 (2 Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]], Ix, Ix, Iy, Iy, Sz, Sz, Sz] +
     Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]],
      Ix, Iy, Ix, Iy, Sz, Sz, Sz] +
     Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]],
      Iy, Ix, Ix, Iy, Sz, Sz, Sz] -
     2 Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]], Iy, Iy, Iy, Iy, Sz, Sz, Sz] -
     Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]],
      Iz, Ix, Ix, Iz, Sz, Sz, Sz] -
     Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]],
      Iz, Ix, Iz, Ix, Sz, Sz, Sz] -
     2 (Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]], Iz, Iz, Ix, Ix, Sz, Sz, Sz] -
        Mult[Inv[Mult[Iy, Iy, Sz] - Mult[Iz, Iz, Sz]], Iz, Iz, Iz, Iz, Sz, Sz, Sz]))}}
```

Harmonic oscillator test case

In[176]:=
```
Clear[aL$sym, aR$sym, ω, Q$sym, P$sym, H$sym, Q, P];
CreateScalar[ω, delta$x$sym, delta$p$sym];
OscSingle$CreateOperators[aL$sym, aR$sym];
```

OscSingle$CreateOperators: Creating oscillator operators.

OscSingle$CreateOperators: Adding oscillator commutations relations.

In[179]:=
```
H$sym = ω (Mult[aL$sym, aR$sym] + Mult[aR$sym, aL$sym]) / 2;
```

In[180]:=
```
CreateOperator[{{Q, P}}];
{Q$sym, P$sym} = {(aR$sym + aL$sym) / Sqrt[2], I (aR$sym - aL$sym) / Sqrt[2]};
QP$rules = {aR$sym → (Q - I P) / Sqrt[2], aL$sym → (Q + I P) / Sqrt[2]};
```

Check that we recover the expected Hamiltonian written in terms if (P,Q).

In[183]:=
```
H$sym /. QP$rules // Simplify
```

Out[183]= $\frac{1}{2}\,\omega\,(\texttt{Mult}[P, P] + \texttt{Mult}[Q, Q])$

Here is a test of **Evolver[]** from Evolve-tests.m. It gives the expected result.

In[184]:=
```
ExpToTrig[Expand[Evolver[H$sym, t$sym, Q$sym]]] /. QP$rules // Simplify
```

Out[184]= Q Cos[t$sym ω] - P Sin[t$sym ω]

Here is the same test applied to **Evolver2[]**. It also gives the expected result.

In[185]:=
```
ExpToTrig[Expand[Evolver2[H$sym, t$sym, Q$sym]]] /. QP$rules // Simplify
```

Out[185]= Q Cos[t$sym ω] - P Sin[t$sym ω]

Check that Evolver2[] evolves the raising and lower operators correctly. It does.

In[186]:=
```
{Evolver2[H$sym, t$sym, aR$sym], Evolver2[H$sym, t$sym, aL$sym]}
```

Out[186]= $\{\text{aR\$sym}\, e^{-i\, \text{t\$sym}\, \omega}, \text{aL\$sym}\, e^{i\, \text{t\$sym}\, \omega}\}$

In the following cell we use Evolve[] expand Q$sym into a sum of two operators, evolve these operators separately under the Hamiltonian using Evolver2[], add the results together, and convert raising and lower operators into (Q,P) operators.   This procedure also gives the correct result.

In[187]:=
```
Evolve[H$sym, t$sym, Q$sym] /. Evolve → Evolver2 /. QP$rules  // FullSimplify
```

Out[187]= Q Cos[t$sym ω] - P Sin[t$sym ω]

Let's dig a little deeper into how Evolver2[] gets the right answer. We see that the 0th and 2nd commutators are proportional. The algorithm correctly identifies this as a 2E case. However, when it tries to “divide” the operators, it cannot simplify the resulting expressions.

In[188]:=
```
Evolver2[H$sym, t$sym, Q$sym, quiet → False]
```

$$\rho^{(n)} = \left\{\frac{\text{aL\$sym} + \text{aR\$sym}}{\sqrt{2}}, \frac{i\,(\text{aL\$sym} - \text{aR\$sym})\,\omega}{\sqrt{2}}, -\frac{(\text{aL\$sym} + \text{aR\$sym})\,\omega^2}{\sqrt{2}}, -\frac{i\,(\text{aL\$sym} - \text{aR\$sym})\,\omega^3}{\sqrt{2}}\right\}$$

$$\text{commutators} = \left\{\frac{i\,(\text{aL\$sym} - \text{aR\$sym})\,\omega}{\sqrt{2}}, -i\,\omega, 0, 0\right\}$$

$$\begin{aligned}\text{divisions} = \{ & i\,\omega\,(\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] - \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]), \\ & -\omega^2\,(\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]), \\ & -\omega^2\,(\text{Mult}[\text{Inv}[\text{aL\$sym} - \text{aR\$sym}], \text{aL\$sym}] - \text{Mult}[\text{Inv}[\text{aL\$sym} - \text{aR\$sym}], \text{aR\$sym}])\}\end{aligned}$$

Case 2E

$$w = \omega\,\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}$$

Out[188]=

$$\begin{aligned}&\frac{1}{\sqrt{2}}\Big(\text{Mult}\Big[\text{Cos}\Big[\text{t\$sym}\,\omega\,\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}\Big], \\ &\quad \text{aL\$sym}\Big] + \text{Mult}\Big[\text{Cos}\Big[\text{t\$sym}\,\omega\,\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}\Big], \\ &\quad \text{aR\$sym}\Big]\Big) + \frac{1}{\sqrt{2}}\,i\,\Big(\text{Mult}\Big[\text{Sin}\Big[\text{t\$sym}\,\omega\,\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}\Big], \\ &\quad \text{aL\$sym}, \frac{1}{\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}}\Big] - \text{Mult}\Big[\\ &\quad \text{Sin}\Big[\text{t\$sym}\,\omega\,\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}\Big], \\ &\quad \text{aR\$sym}, \frac{1}{\sqrt{\text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aL\$sym}] + \text{Mult}[\text{Inv}[\text{aL\$sym} + \text{aR\$sym}], \text{aR\$sym}]}}\Big]\Big)\end{aligned}$$

In the next step we write the raising and lowering operators in terms of (Q, P).

In[189]:=

```
Evolver2[H$sym, t$sym, Q$sym] /. QP$rules
```

Out[189]=

$$\begin{aligned}&\frac{1}{\sqrt{2}} \\ &\Bigg(\frac{1}{\sqrt{2}}\Bigg(-i\,\text{Mult}\Bigg[\text{Cos}\Bigg[\text{t\$sym}\,\omega\,\sqrt{\Bigg(\frac{-i\,\text{Mult}\Big[\text{Inv}\Big[\frac{-i\,P+Q}{\sqrt{2}} + \frac{i\,P+Q}{\sqrt{2}}\Big], P\Big] + \text{Mult}\Big[\text{Inv}\Big[\frac{-i\,P+Q}{\sqrt{2}} + \frac{i\,P+Q}{\sqrt{2}}\Big], Q\Big]}{\sqrt{2}} +} \\ &\qquad \overline{\frac{i\,\text{Mult}\Big[\text{Inv}\Big[\frac{-i\,P+Q}{\sqrt{2}} + \frac{i\,P+Q}{\sqrt{2}}\Big], P\Big] + \text{Mult}\Big[\text{Inv}\Big[\frac{-i\,P+Q}{\sqrt{2}} + \frac{i\,P+Q}{\sqrt{2}}\Big], Q\Big]}{\sqrt{2}}\Bigg)}\Bigg], P\Bigg] +\end{aligned}$$

$$\text{Mult}\Big[\text{Cos}\Big[\text{t\$sym}\,\omega\,\sqrt{\left(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}\right)}\Big],Q\Big]\Big)+\frac{1}{\sqrt{2}}$$

$$\Big(i\,\text{Mult}\Big[\text{Cos}\Big[\text{t\$sym}\,\omega\,\sqrt{\left(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}\right)}\Big],P\Big]+$$

$$\text{Mult}\Big[\text{Cos}\Big[\text{t\$sym}\,\omega\,\sqrt{\left(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}\right)}\Big],$$

$$Q\Big]\Big)\Big)+\frac{1}{\sqrt{2}}\,i\,\Big(-\frac{1}{\sqrt{2}}\Big(-i\,\text{Mult}\Big[\text{Sin}\Big[\text{t\$sym}\,\omega$$

$$\sqrt{\left(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}\right)}\Big],P,$$

$$1\Big/\left(\sqrt{\left(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}\right)}\right)\Big]+$$

$$\text{Mult}\Big[\text{Sin}\Big[\text{t\$sym}\,\omega\,\sqrt{\Big(\frac{-i\,\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\,P+Q}{\sqrt{2}}+\frac{i\,P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+}$$

$$\left.\left.\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], Q\right]}{\sqrt{2}}\right)\right], Q,$$

$$\frac{1}{\sqrt{\frac{-i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}}}$$

$$\left.\left.\right]\right)+\frac{1}{\sqrt{2}}\left(i\, \text{Mult}\left[\text{Sin}\left[\text{t\$sym}\, \omega\right.\right.\right.$$

$$\sqrt{\left(\frac{-i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], Q\right]}{\sqrt{2}}+\right.}$$

$$\left.\left.\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], Q\right]}{\sqrt{2}}\right)\right], P,$$

$$\frac{1}{\sqrt{\frac{-i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}}}$$

$$\left.\right]+\text{Mult}\left[\text{Sin}\left[\text{t\$sym}\, \omega\right.\right.$$

$$\sqrt{\left(\frac{-i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], Q\right]}{\sqrt{2}}+\right.}$$

$$\left.\left.\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right], Q\right]}{\sqrt{2}}\right)\right], Q,$$

$$\left.\frac{1}{\sqrt{\frac{-i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}+\frac{i\, \text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],P\right]+\text{Mult}\left[\text{Inv}\left[\frac{-i\, P+Q}{\sqrt{2}}+\frac{i\, P+Q}{\sqrt{2}}\right],Q\right]}{\sqrt{2}}}}\right]$$

$$\left.\left.\right)\right)$$

Amazingly, this simplifies to the expected result.

In[190]:=
```
Evolver2[H$sym, t$sym, Q$sym] /. QP$rules  // Simplify
```

Out[190]=
```
Q Cos[t$sym ω] - P Sin[t$sym ω]
```