
The UniDyn *Mathematica* package: Self-derived unitary operator rotations in quantum mechanics

John A. Marohn*

Dept. of Chemistry and Chemical Biology
Cornell University, Ithaca, NY 14851-1301, USA

December 25, 2025

Abstract

We have developed a *Mathematica* algorithm for symbolically calculating unitary transformations of quantum-mechanical operators. The algorithm obtains closed-form analytical results, does not rely on a matrix representation of operators, and is applicable to both bounded systems like coupled spins and unbounded systems like harmonic oscillators. The rotations are *self derived* from the operators' underlying commutation relations and can be carried out in a *basis free* way. Example calculations are presented involving magnetic resonance and quantum optics.

1 Introduction

Given a time-independent Hamiltonian \mathcal{H} and an initial (density) operator $\rho(0)$, the `Evolver` algorithm described below computes the (density) operator at time t by implementing the following unitary rotation

$$\rho(t) = e^{-i\mathcal{H}t} \rho(0) e^{+i\mathcal{H}t} \equiv \text{Evolver}[\mathcal{H}, t, \rho(0)] \quad (1)$$

Consider two representative examples. The time evolution of transverse spin magnetization in a longitudinal magnetic field is described by the rotation

$$e^{-i\omega t I_z} I_x e^{+i\omega t I_z} = I_x \sin(\omega t) + I_y \cos(\omega t) \quad (2)$$

* jam99@cornell.edu

where ω is the Larmor frequency and I_x , I_y , and I_z are the spin angular-momentum operators. The harmonic oscillator's creation operator a^\dagger evolves in time under the harmonic oscillator's Hamiltonian as follows:

$$e^{-i\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)} a^\dagger e^{+i\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)} = a^\dagger e^{-i\omega t} \quad (3)$$

where a is the annihilation operator and ω is the harmonic-oscillator's resonance frequency. Knowing the commutation relations among the operators comprising \mathcal{H} and $\rho(0)$, it is often possible to obtain a closed-form algebraic solution to Eq. 1. The `Evolver` algorithm obtains a closed-form solution to Eq. 1 by extending an approach introduced by Slichter in Section 2.3 of his *Principles of Magnetic Resonance* text [1]. The only input to the algorithm is the commutation relations for the relevant operators.

2 Background

Numerous software programs exist for computing eq. 1 dynamics numerically. In order to gain physical understanding, researchers want to compute the dynamics symbolically, using computer algebra programs. This effort has been motivated by applications in magnetic resonance [2–10], quantum computing [11–13], electronic structure [14], vibrational dynamics [15], and quantum optics [16, 17]. Unitary rotations like eq. 1 occur in exact effective Hamiltonian theory [18, 19], the Baker-Campbell-Hausdorff formula [15, 20], the Zassenhaus product formula [15, 20, 21], the Hadamard lemma [15], and simplifications of exponential operators involving raising and lower operators that are possible if the underlying potential has certain symmetries [15].

Most eq. 1 work to date, motivated by applications in magnetic resonance and quantum computing, has been semi-symbolic. Spins and few-level electron systems have a finite Hilbert space and can be represented by finite-dimensional matrices. In the semi-symbolic approaches, spin-angular momentum operators are expressed as matrices [5, 6, 11–13, 23] or irreducible spherical tensors [8, 9, 22, 23], while Hamiltonian and irradiation parameters remain symbolic. This semi-symbolic approach has led to powerful *Mathematica* packages like *MathNMR* [22] and *SpinDynamica* [23].

Vibrations and photons have an infinite (or near-infinite) Hilbert space and consequently the associated creation and annihilation operators do not have a matrix representation. Evolving creation and annihilation operators requires purely symbolic approaches.

There are fewer examples of purely symbolic approaches to computing eq. 1 dynamics. In magnetic resonance, spin angular momenta have been

described using non-commuting product operators, with evolution under pulsed irradiation and free evolution implemented either approximately, using a Baker-Campbell-Hausdorff (BCH) expansion [10], or exactly using operator substitution rules [2–4, 7]. Computer algebra programs have been used to carry out the multiplication and commutation of harmonic-oscillator (*i.e.* boson) operators symbolically [14, 16, 17], but eq. 1 dynamics was only computed approximately using a BCH expansion, with the unitary transformation represented by a truncated, finite series of nested commutators. Nguyen and workers described a promising procedure for resumming the resulting finite series to give a closed-form expression [17]. We found that this approach fails when the arguments of the associated sines, cosines, and exponentials involve non-trivial expressions such as fractions. We know of only one example, Ref. 14, which attempted to calculate the joint evolution of spin- and harmonic-oscillator operators in a purely symbolic way.

Slichter was interested in calculating the unitary evolution of angular momentum operators arising in spin-physics problems [1], like the rotation in eq. 2. In the following section we introduce the Slichter procedure by considering example cases. We show that his procedure works well for harmonic oscillator problems as well, including problems involving the coupling of a harmonic oscillator with a two-level system, a minimum model for describing electron transfer in chemical reactions. It will become apparent that automating Slichter’s procedure in a computer algebra program like *Mathematica* is possible but challenging. We introduce a generalization of the Slichter algorithm that is well suited for automation by a computer algebra program. We have implemented this algorithm as a *Mathematica* function, `Evolver`. This function evaluates Eq. 1 given an \mathcal{H} , an initial operator $\rho(0)$, and the commutation relations between the operators comprising \mathcal{H} and $\rho(0)$.

3 The algorithm

To understand the `Evolver` algorithm it is essential to understand Slichter’s approach to calculating the unitary evolution of an operator. To help the reader understand and appreciate Slichter’s approach, let us briefly review two methods commonly used for solving Eq. 1. As an example, consider the case where $\mathcal{H} = \omega I_z$ and $\rho(0) = I_+ = I_x + i I_y$. In this case we want to compute

$$\rho(t) = e^{-i \omega t I_z} I_+ e^{+i \omega t I_z}. \quad (4)$$

Method 1 — One approach to computing $\rho(t)$ is to expand Eq. 4 in a Taylor series,

$$\rho(t) = \rho(0) + \rho^{(1)}(0)t + \frac{1}{2!}\rho^{(2)}(0)t^2 + \frac{1}{3!}\rho^{(3)}(0)t^3 + \dots \quad (5)$$

The coefficients are obtained by differentiating ρ and setting $t \rightarrow 0$. Taking the first derivative,

$$\rho^{(1)}(t) = e^{-i\omega t I_z} (-i\omega I_z) I_+ e^{+i\omega t I_z} + e^{-i\omega t I_z} I_+ (+i\omega I_z) e^{+i\omega t I_z} \quad (6a)$$

$$= e^{-i\omega t I_z} (-i\omega [I_z, I_+]) e^{+i\omega t I_z} \quad (6b)$$

$$\rho^{(1)}(0) = -i\omega I_+ \quad (6c)$$

where $\rho^{(n)}$ represents the n^{th} derivative with respect to time and where we have used $[I_z, I_+] = I_+$ to simplify the commutator. Taking the second derivative,

$$\rho^{(2)}(t) = e^{-i\omega t I_z} ((-i\omega)^2 [I_z, [I_z, I_+]]) e^{+i\omega t I_z} \quad (7a)$$

$$\rho^{(2)}(0) = (-i\omega)^2 I_+ \quad (7b)$$

By induction, we see that

$$\rho^{(n)}(0) = (-i\omega)^n I_+ \quad (8)$$

Substituting this finding into the Taylor expansion gives

$$\rho(t) = I_+ \left(1 + (-i\omega t) + \frac{1}{2!}(-i\omega t)^2 + \frac{1}{3!}(-i\omega t)^3 + \dots \right) \quad (9)$$

We are now supposed to recognize the term in parenthesis as the Taylor series of $e^{-i\omega t}$. This insight enables us to resum the infinite series in the Taylor expansion to obtain the closed-form result

$$\rho(t) = I_+ e^{-i\omega t}. \quad (10)$$

Method 2 — A second approach to evaluating Eq. 4 is to expand the exponential using the Löwdin projection-operator theorem [24]. This theorem allows us to expand a function of an operator — I_z here — in terms involving the function evaluated at the operator's eigenvalues times an operator that project's onto the eigenvalue's subspace. Taking the total spin angular momentum to be $I = 1/2$ for simplicity, the relevant eigenvalues are $+1/2$ and $-1/2$ and the relevant projection operators are $\mathcal{P}_{1/2} = |\alpha\rangle\langle\alpha|$ and $\mathcal{P}_{-1/2} = |\beta\rangle\langle\beta|$. Applying Löwdin's theorem,

$$e^{-i\omega t I_z} = e^{-i\omega t/2} |\alpha\rangle\langle\alpha| + e^{+i\omega t/2} |\beta\rangle\langle\beta|. \quad (11)$$

Substituting this result into Eq. 4 yields

$$\rho(t) = \left(e^{-i\omega t/2} |\alpha\rangle \langle\alpha| + e^{+i\omega t/2} |\beta\rangle \langle\beta| \right) I_+ \left(e^{+i\omega t/2} |\alpha\rangle \langle\alpha| + e^{-i\omega t/2} |\beta\rangle \langle\beta| \right)$$

Applying the relations $I_+ |\alpha\rangle = 0$, $I_+ |\beta\rangle = |\alpha\rangle$, $\langle\alpha|\alpha\rangle = 1$, $\langle\beta|\alpha\rangle = 0$, and $|\alpha\rangle \langle\beta| = I_+$, this expression simplifies to

$$\rho(t) = I_+ e^{-i\omega t}. \quad (12)$$

While both these approaches yield closed-form solutions, each is hardly extensible. The first method requires the resummation of a Taylor series; this step would be difficult or impossible to automate for any evolution more complicated than the one above. The second method requires obtaining the eigenvalues of the Hamiltonian, usually by reducing it to matrix form and diagonalizing it. It is hard to see how to apply this diagonalization procedure in an infinite-level system like the idealized harmonic oscillator. Even for a spin problem where the number of levels is finite, diagonalizing \mathcal{H} would force us to write down the matrix representation of \mathcal{H} which would in turn commit us to specifying the total angular momentum I of the spin we are interested in. In many problems, we would like to obtain a solution valid for a spin of *any* I .

Method 3, Example 1 — Now consider Slichter's procedure. Let us take another look at the derivative of ρ :

$$\dot{\rho}(t) = e^{-i\omega t I_z} (-i\omega [I_z, I_+]) e^{+i\omega t I_z} \quad (13a)$$

$$= -i\omega \left(e^{-i\omega t I_z} I_+ e^{+i\omega t I_z} \right) \quad (13b)$$

As before we have used $[I_z, I_+] = I_+$ to reduce the commutator in Eq. 13a. The key insight in the Slichter procedure is that the term in parenthesis in Eq. 13b is nothing more than the original time dependent density operator, $\rho(t)$. This insight allows us to write Equation 13b as

$$\dot{\rho}(t) = -i\omega \rho(t) \quad (14)$$

In this differential equation, ω is a *number*, while $\rho(t)$ is an *operator*. The solution to this differential equation is

$$\rho(t) = \rho(0) e^{-i\omega t} = I_+ e^{-i\omega t}, \quad (15)$$

which is easily verified by back substitution.

Example 2 — An analogous calculation arises in a harmonic oscillator problem where the Hamiltonian is $\mathcal{H} = \omega(a^\dagger a + 1/2)$ and the initial operator is $\rho(0) = a^\dagger$:

$$\rho(t) = e^{-i\omega t(a^\dagger a + 1/2)} a^\dagger e^{+i\omega t(a^\dagger a + 1/2)} \quad (16)$$

Using the same procedure and the commutation relation $[a^\dagger a, a^\dagger] = a^\dagger[a, a^\dagger] + [a^\dagger, a^\dagger]a = a^\dagger$, this equation reduces to

$$\rho(t) = a^\dagger e^{-i\omega t}. \quad (17)$$

These first two Slichter-procedure example cases have in common that the commutator of the Hamiltonian with the operator of interest is simply proportional to the operator. As a result of this underlying commutation relation, the problem of calculating Eq. 1 in both cases has been reduced to the problem of solving a first-order differential equation, Eq. 14. In light of the two previous methods, the Slichter procedure is rather remarkable. It allows us to obtain a closed-form solution for Eq. 4 without resorting to Taylor series and without even requiring knowledge of the Hamiltonian's eigenvalues.

Example 3 — The Slichter procedure is readily applied to more complicated unitary-evolution problems. Consider the case where $\mathcal{H} = \omega I_z$ and $\rho(0) = I_x$. Then

$$\rho(t) = e^{-i\omega t I_z} I_x e^{+i\omega t I_z}. \quad (18)$$

Taking the time derivative we obtain

$$\dot{\rho}(t) = e^{-i\omega t I_z} (-i\omega [I_z, I_x]) e^{+i\omega t I_z} \quad (19a)$$

$$= \omega \left(e^{-i\omega t I_z} I_y e^{+i\omega t I_z} \right) \quad (19b)$$

where we have used $[I_z, I_x] = i I_y$ to reduce the commutator in Eq. 19a. In contrast to the previous case, $\dot{\rho}(t)$ is not proportional to ρ . Taking another time derivative we obtain

$$\ddot{\rho}(t) = \omega e^{-i\omega t I_z} (-i\omega [I_z, I_y]) e^{+i\omega t I_z} \quad (20a)$$

$$= -\omega^2 \left(e^{-i\omega t I_z} I_x e^{+i\omega t I_z} \right) \quad (20b)$$

where we have used $[I_z, I_y] = -i I_x$ to reduce the commutator in Eq. 19a. The term in parenthesis in Eq. 20b is proportional to $\rho(t)$ and consequently

$$\ddot{\rho}(t) = -\omega^2 \rho(t). \quad (21)$$

The solution to this second-order differential equation is

$$\rho(t) = \rho(0) \cos(\omega t) + \frac{\dot{\rho}(0)}{\omega} \sin(\omega t). \quad (22)$$

We are given that $\rho(0) = I_x$ and we see from Eq. 19b that $\dot{\rho}(0) = \omega I_y$. Plugging these initial conditions into the above equation we obtain

$$\rho(t) = I_x \cos(\omega t) + I_y \sin(\omega t) \quad (23)$$

as the solution to Eq. 18.

We can imagine automating the steps leading to Eqs. 15, 16, and 23. What is required is the ability to

1. perform non-commutative algebra,
2. evaluate commutators, and
3. determine which differential equation an evolved operator satisfies.

Mathematica has a native function for carrying out non-commutative multiplication, but this function has essentially no simplification rules associated with it; for example,

```
In[1] := a ** (2 b) // Simplify
Out[1] = a ** (2 b)
```

This limitation was resolved by Helton and co-workers, whose NCALgebra package [25] gives *Mathematica* the ability to manipulate non-commuting algebraic expressions. This package allows the user to define variables as either commuting (e.g., ω) or non-commuting (e.g., I_+ , I_x , a^\dagger , and so on). This package forms an excellent starting point for implementing non-commutative multiplication, and expanding and simplifying commutators. We find, however, that using the package adds significant computational overhead, increasing computation time over ten-fold. In the code below, we instead introduce our own non-commutative multiplication function and accompanying simplification rules.

Based on our experience so far, developing an algorithm capable of self-deriving Eqs. 15, 16, and 23 would now seem to be a matter of calculating

$$\rho^{(1)} = U^\dagger [-i\mathcal{H}, \rho(0)] U \quad (24)$$

$$\rho^{(2)} = U^\dagger [-i\mathcal{H}, [-i\mathcal{H}, \rho(0)]] U = U^\dagger [-i\mathcal{H}, \rho^{(1)}] U \quad (25)$$

and continuing until we obtain an expression that is proportional to $\rho = U^\dagger \rho(0) U$. If n iterations are required, then $\rho(t)$ must satisfy an n^{th} order differential equation. If the equation's coefficients and initial conditions can

be extracted from the available non-commutative expressions correctly, then the differential equation can be fed to *Mathematica* to solve. This is roughly the procedure we will use. Because it would have to handle solving a large number of possible differential equations, however, significant effort would be required to implement the procedure just outlined in an automated way. The next example highlights why this is so. This example will serve as our launching point for introducing a revised, simple, and general algorithm for implementing the Slichter procedure for evaluating unitary evolution in an automated way.

Example 4 — Consider a unitary evolution with $\mathcal{H} = \Delta\omega I_z + \omega_1 I_x$ and $\rho(0) = I_z$. This rotation is involved in calculating the evolution of the difference in populations of a two level system when near-resonant irradiation is applied. The two-level system could be, for example, a spin $I = 1/2$ particle like a proton spin in a magnetic resonance experiment or it could be the lowest two electronic energy levels of an atom in a quantum optics experiment. The unitary evolution we are interested in computing is

$$\rho(t) = e^{-i(\Delta\omega I_z + \omega_1 I_x)t} I_z e^{+i(\Delta\omega I_z + \omega_1 I_x)t} \equiv U^\dagger I_x U \quad (26)$$

Computing the first few derivatives, we find

$$\dot{\rho} = U^\dagger (-i[\Delta\omega I_z + \omega_1 I_x, I_z]) U \quad (27a)$$

$$= U^\dagger (-\omega_1 I_y) U \quad (27b)$$

$$\ddot{\rho} = U^\dagger (-i[\Delta\omega I_z + \omega_1 I_x, -\omega_1 I_y]) U \quad (27c)$$

$$= U^\dagger (\Delta\omega \omega_1 I_x - \omega_1^2 I_z) U \quad (27d)$$

$$\ddot{\rho} = U^\dagger (-i[\Delta\omega I_z + \omega_1 I_x, \Delta\omega \omega_1 I_x - \omega_1^2 I_z]) U \quad (27e)$$

$$= U^\dagger (\omega_1 (\omega_1^2 + \Delta\omega^2) I_y) U \quad (27f)$$

Neither $\dot{\rho}$ nor $\ddot{\rho}$ is proportional to ρ . We see, however, that $\ddot{\rho}$ is proportional to $\dot{\rho}$. Defining $\sigma \equiv \dot{\rho}$, we see that σ satisfies the following differential equation

$$\ddot{\sigma} = -(\Delta\omega^2 + \omega_1^2) \sigma \quad (28)$$

with initial conditions

$$\sigma(0) = \dot{\rho}(0) = -\omega_1 I_y \quad (29a)$$

$$\dot{\sigma}(0) = \ddot{\rho}(0) = \Delta\omega \omega_1 I_x - \omega_1^2 I_z \quad (29b)$$

The solution to Eq. 28 is

$$\sigma(t) = \sigma(0) \cos(\omega_{\text{eff}} t) + \frac{\dot{\sigma}(0)}{\omega_{\text{eff}}} \sin(\omega_{\text{eff}} t) \quad (30)$$

with

$$\omega_{\text{eff}} \equiv \sqrt{\Delta\omega^2 + \omega_1^2} \quad (31)$$

an effective evolution frequency. Plugging in initial conditions,

$$\sigma(t) = \frac{\Delta\omega \omega_1}{\omega_{\text{eff}}} I_x \sin(\omega_{\text{eff}} t) - \omega_1 I_y \cos(\omega_{\text{eff}} t) - \frac{\omega_1^2}{\omega_{\text{eff}}} I_z \sin(\omega_{\text{eff}} t) \quad (32)$$

To obtain an equation for $\rho(t)$ we need to integrate this equation for $\sigma(t)$, taking care to include a constant of integration. Noting that

$$\int \cos(\omega_{\text{eff}} t) dt = \frac{\sin(\omega_{\text{eff}} t)}{\omega_{\text{eff}}} \text{ and } \int \sin(\omega_{\text{eff}} t) dt = -\frac{\cos(\omega_{\text{eff}} t)}{\omega_{\text{eff}}},$$

the solution for $\rho(t)$ becomes

$$\rho(t) = -\frac{\Delta\omega \omega_1}{\omega_{\text{eff}}^2} I_x \sin(\omega_{\text{eff}} t) - \frac{\omega_1}{\omega_{\text{eff}}} I_y \sin(\omega_{\text{eff}} t) + \frac{\omega_1^2}{\omega_{\text{eff}}^2} I_z \cos(\omega_{\text{eff}} t) + c \quad (33)$$

The integration constant c is determined by requiring the above equation to satisfy $\rho(0) = I_z$, the initial condition. Solving for the integration constant,

$$c = \frac{\Delta\omega \omega_1}{\Delta\omega^2 + \omega_1^2} I_x + \frac{\Delta\omega^2}{\Delta\omega^2 + \omega_1^2} I_z \quad (34)$$

Plugging this integration constant into Eq. 33 we obtain the following solution for the density operator in Eq. 26:

$$\begin{aligned} \rho(t) = & \left(\frac{\Delta\omega^2}{\Delta\omega^2 + \omega_1^2} I_z + \frac{\Delta\omega \omega_1}{\Delta\omega^2 + \omega_1^2} I_x \right) + \frac{\omega_1}{\sqrt{\Delta\omega^2 + \omega_1^2}} \sin(\sqrt{\Delta\omega^2 + \omega_1^2} t) \\ & + \left(\frac{\omega_1^2}{\Delta\omega^2 + \omega_1^2} I_z - \frac{\Delta\omega \omega_1}{\Delta\omega^2 + \omega_1^2} I_x \right) \cos(\sqrt{\Delta\omega^2 + \omega_1^2} t). \end{aligned} \quad (35)$$

We see in this example an answer for $\rho(t)$ that is markedly more complicated than in the previous examples, the solution to (essentially) a third order differential equation. To uncover this differential equation, it was necessary to compare subsequent derivatives of $\rho(t)$ not to $\rho(0)$ but to $\dot{\rho}(0)$ instead.

Considering in total the four examples of unitary evolution considered so far, it would seem that any algorithm we develop needs to consider the possibility that the density operator satisfies a first, second, or even third-order differential equation.

Method 4 — We now show by example that the single differential equation in each of these four cases can be reduced to a set of *coupled first-order* differential equations — a significant step towards making a simple automated, unitary evolution algorithm. We would expect the four cases to each still require a decision on the number of coupled equations required to solve for ρ in each case. Surprisingly, no decision is needed. We will show that all the cases we have discussed so far can be handled by a set of four coupled differential equations. This is the essential new insight implemented here in the `Evolver` algorithm.

To understand the new method, consider transforming the second-order differential equation in Eq. 21 into two coupled first-order equations. Let the two new variables be

$$x_1 = \rho \text{ and } x_2 = \dot{\rho}. \quad (36)$$

Taking the time derivative of each of these variables we obtain

$$\dot{x}_1 = \dot{\rho} = x_2, \text{ and} \quad (37a)$$

$$\dot{x}_2 = \ddot{\rho} = -\omega^2 \rho = -\omega^2 x_1. \quad (37b)$$

It is apparent that these two variables satisfy the following set of coupled first order equations

$$\frac{d}{dt} \begin{bmatrix} x_2 \\ x_1 \end{bmatrix} = \begin{bmatrix} 0 & -\omega^2 \\ 1 & 0 \end{bmatrix} \begin{bmatrix} x_2 \\ x_1 \end{bmatrix} \text{ with } \begin{bmatrix} x_2(0) \\ x_1(0) \end{bmatrix} = \begin{bmatrix} \omega I_y \\ I_x \end{bmatrix}. \quad (38)$$

Solving this set of coupled equations, we find $x_1(t) = \rho(t) = I_x \sin(\omega t) + I_y \cos(\omega t)$, the expected answer.

What if we did not know how many coupled equations were necessary to solve the problem? Let's consider what would happen if we guessed that three coupled equations were needed instead of two. Defining

$$x_1 = \rho \text{ and } x_2 = \dot{\rho} \text{ and } x_3 = \ddot{\rho}, \quad (39)$$

and taking the time derivative gives

$$\dot{x}_1 = \dot{\rho} = x_2, \quad (40a)$$

$$\dot{x}_2 = \ddot{\rho} = x_3, \text{ and} \quad (40b)$$

$$\dot{x}_3 = \ddot{\rho} \quad (40c)$$

$$= \omega^2 U^\dagger (-i \omega [I_z, I_x]) U \quad (40d)$$

$$= -\omega^3 U^\dagger I_y U \quad (40e)$$

$$= -\omega^2 \dot{\rho} \quad (40f)$$

$$= -\omega^2 x_2. \quad (40g)$$

In writing \dot{x}_3 we have used the shorthand $U(t) \equiv \exp[i\omega t I_z]$ and employed Eq. 19b to simplify the result. The three variables satisfy the following set of coupled first-order differential equations

$$\frac{d}{dt} \begin{bmatrix} x_3 \\ x_2 \\ x_1 \end{bmatrix} = \begin{bmatrix} 0 & -\omega^2 & 0 \\ 1 & 0 & 0 \\ 0 & 1 & 0 \end{bmatrix} \begin{bmatrix} x_3 \\ x_2 \\ x_1 \end{bmatrix} \text{ with } \begin{bmatrix} x_3(0) \\ x_2(0) \\ x_1(0) \end{bmatrix} = \begin{bmatrix} -\omega^2 I_x \\ \omega I_y \\ I_x \end{bmatrix} \quad (41)$$

Solving this new set of three coupled equations gives $x_1(t) = \rho(t) = I_x \sin(\omega t) + I_y \cos(\omega t)$ — the *same answer* that was obtained by solving the set of two coupled equations, Eqs. 38. This finding suggests that we are at liberty to “overguess” the number of equations required to solve the problem.

Let us see how the new method would be applied to solve the Example 4 problem with $\mathcal{H} = \Delta\omega I_z + \omega_1 I_x$ and $\rho(0) = I_z$. The time derivatives up to third order may be summarized as

$$\begin{bmatrix} \rho^{(3)} \\ \rho^{(2)} \\ \rho^{(1)} \\ \rho^{(0)} \end{bmatrix} = U^\dagger \begin{bmatrix} -\Delta\omega(\Delta\omega^2 + \omega_1^2) I_y \\ -\Delta\omega(\Delta\omega I_x - \omega_1 I_z) \\ \Delta\omega I_y \\ I_x \end{bmatrix} U \quad (42)$$

with $\rho^{(n)}$ the n^{th} derivative of ρ with respect to time and

$$U \equiv \exp[i(\Delta\omega I_z + \omega_1 I_x)t]. \quad (43)$$

We see that $\rho^{(3)} = -(\Delta\omega^2 + \omega_1^2)\rho^{(1)}$, as we showed before. Defining $x_n = \rho^{(n-1)}$ for $n = 1$ through 4, we obtain the following set of four coupled equations describing the time evolution of the density operator:

$$\underbrace{\frac{d}{dt} \begin{bmatrix} x_4 \\ x_3 \\ x_2 \\ x_1 \end{bmatrix}}_{\lambda} = \underbrace{\begin{bmatrix} 0 & -(\Delta\omega^2 + \omega_1^2) & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \end{bmatrix}}_{\Omega} \underbrace{\begin{bmatrix} x_4 \\ x_3 \\ x_2 \\ x_1 \end{bmatrix}}_{\lambda} \quad (44)$$

with

$$\underbrace{\begin{bmatrix} x_4(0) \\ x_3(0) \\ x_2(0) \\ x_1(0) \end{bmatrix}}_{\lambda_0} = \begin{bmatrix} -\Delta\omega(\Delta\omega^2 + \omega_1^2) I_y \\ -\Delta\omega(\Delta\omega I_x - \omega_1 I_z) \\ \Delta\omega I_y \\ I_x \end{bmatrix} \quad (45)$$

Solving these coupled equations, we obtain Eq. 35 for $x_1 = \rho(t)$.

In the above equations we have defined λ as the vector of x_n 's we are interested in and Ω as the matrix of coefficients coupling λ to $\dot{\lambda}$. The coupled equations are summarized as

$$\frac{d\lambda}{dt} - \Omega \cdot \lambda = 0 \quad (46)$$

with λ_0 given. The `Evolver` algorithm proceeds as follows.

1. Evaluate the derivatives $\rho^{(1)}$ through $\rho^{(3)}$ by repeatedly applying computing $\rho^{(n+1)} = [-iH, \rho^{(n)}]$, starting from $\rho^{(0)} = \rho(0)$. Assign the initial-condition vector $\lambda_0 = (\rho^{(3)}, \dots, \rho^{(0)})$.
2. Fill in the lower 3 rows of Ω by placing a 1 in entries one below the diagonal and 0 in all the other positions.
3. Look for an entry in the list $(\rho^{(2)}, \rho^{(1)}, \rho^{(0)})$ that is proportional to $\rho^{(3)}$, starting with the $\rho^{(2)}$ (entry $n = 3$) and working towards $\rho^{(0)}$ (entry $n = 1$). Call the entry where the match occurs n_{match} . Call the ratio between $\rho^{(3)}$ and the matching entry r .
4. Fill in the upper row of Ω with zeros. Assign the value r to the entry in the n_{match} column of the upper row of Ω , counting from the right to the left.
5. Feed Eq. 46, including the initial condition, to *Mathematica* to solve. The time-dependent density operator is $\rho(t) = \lambda_1(t)$.

4 The code

The code consists of `.m` files and *Mathematica* notebooks. Each modular `.m` file implements a desired function, like non-commutative multiplication or commutation. Associated with each `.m` file is a second `.m` file containing *units tests*. The code, at the time of writing, includes over 120 unit tests covering all the functions defined below. More detailed applications of the code are given in separate *Mathematica* notebooks and will be discussed in the next section.

Code was written in the literate-programming style, with L^AT_EX-style text, equations, and tables interspersed with code in the form of commented-out text. In this section we use L^AT_EX's `lstinputlisting` package to import the code and typeset it nicely. This package also takes care of importing and un-commenting the embedded comments so they can be processed in L^AT_EX. The package moreover implements a long list of character substitutions,

allowing the code listings below to have variables appear with subscripts and superscripts.

4.1 Operators and scalars

Our first task is to define functions that enable *Mathematica* to distinguish between (non-commutative) operators and (commutative) scalars. This is done in the `OpQ.m` package, whose listing appears below. A variable is either a commutative *scalar* or non-commutative *operator*. Operators have a *phylum* and an *order*. Operators of different phyla commute. The operator order is used for sorting.

Listing 1: unidyn/OpQ.m

A *simple operator* is a symbol whose upvalue for `SimpleOperatorQ[]` is defined to be **True**. The default return for `SimpleOperatorQ` is **False**, with the result that unless defined otherwise, all arguments to the query `SimpleOperatorQ[]` return **False**. By default then, all quantities are scalars.

```
Clear[SimpleOperatorQ]
SimpleOperatorQ[x_] := False;
```

An *operator* is any expression one of whose atoms is a simple operator. The function call to `Level` in `OperatorQ[]` remarkably pulls out a list of all symbols used in the expression *x* (all its *atoms*.) The function `ScalarQ[]` tests if its argument is a *scalar*. By scalar we mean *not an operator*. As the default return for `OperatorQ[]` is **False**, the default return for `ScalarQ[]` is **True**; all quantities default to scalars.

```
Clear[OperatorQ, ScalarQ]
OperatorQ[x_] :=
  Apply[Or,
    Map[SimpleOperatorQ,
      Level[x, {-1}]]];
ScalarQ[x_] := !OperatorQ[x];
```

You “create” an operator by defining an upvalue for it, e.g., arranging so that `SimpleOperator[the operator]` returns **True**. By using upvalues, you associate this definition not with the function `SimpleOperator[]`, but with the operator itself. This makes for faster computations. The added definitions make it easy to create many operators at once by passing multiple variables or a list of variables to the function `CreateOperator[]`.

```
Clear[CreateOperator];
CreateOperator[a0_Symbol] :=
  (Clear[a0];
   SimpleOperatorQ[a0] ^= True)
CreateOperator[a0_, b0_] :=
  (CreateOperator[a0];
   CreateOperator[b0];)
CreateOperator[a0_?VectorQ] :=
```

```
(CreateOperator /@ a0;)
```

You don't really need to "create" a scalar, since this is the default category for any symbol, given the above definitions. Nevertheless, by defining the upvalue of `ScalarQ` to be `True`, you can speed up computations which involve testing to see whether or not an object is a scalar.

```
Clear[CreateScalar];
CreateScalar[a0_Symbol] :=
  (Clear[a0];
   ScalarQ[a0] ^= True)
CreateScalar[a0_List] :=
  (CreateScalar /@ a0;)
CreateScalar[a0_, b0_] :=
  (CreateScalar[a0];
   CreateScalar[b0];)
```

Passing a matrix to `SimpleOperator` invokes the following function call. This function assigns the operators in the matrix a *phylum* and an *order* which is determined by the operators location in the matrix.

```
CreateOperator[a0_?ListQ] :=
Module{val, m, n},
  (Clear[#]; SimpleOperatorQ[#] ^= True) &
  /@ Flatten[a0];
Do[
  Do[val = a0[[m]][[n]];
    phylum[val] ^= m;
    order[val] ^= n,
    {n, Dimensions[a0[[m]][[1]]}],
  {m, Dimensions[a0][[1]]}
]
```

The idea of an operator having a *phylum* and an *order* can be understood best with an example. Consider, for example, the function call

```
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}]
```

This call will create six operators which, when passed to the functions `phylum` and `order`, will return the following values:

Op	phylum[Op]	order[Op]
I_x	1	1
I_y	1	2
I_z	1	3
S_x	2	1
S_y	2	2
S_z	2	3

Having a `phylum` and an `order` assigned to each operator will be used below to sort operators and to decide whether or not two operators commute.

Unit tests for the functions in the `OpQ.m` package are organized into a separate package, `OpQ-tests.m`, whose listing follows. Unit testing is

built on *Mathematica*'s `VerificationTest` function. When the return value of the `VerificationTest` function is passed to the `TestReport` in a *Mathematica* notebook, the notebook displays a nice graphical report card.

Listing 2: unidyn/OpQ-tests.m

Create a shorthand function for creating unit tests.

```
If [$VersionNumber < 10.,

  vtest[label_, test_] :=
    If [test === True,
      Print[" Pass"],
      Print[" Fail > ", StringJoin["OpQ > test", ToString[label ]]],

  vtest[label_, test_] :=
    VerificationTest [ test,
      True,
      TestID → StringJoin[
        "OpQ > test",
        ToString[label ]]
]
```

Any variable will test **True** when queried by `ScalarQ[]`, whether it has been defined as a scalar or not.

```
vtest [" 1", ScalarQ[a0] == True]
vtest [" 2", CreateScalar[b0];
  ScalarQ[b0] == True]
```

For `OperatorQ[]` to return **True**, in contrast, the variable must have been created as an operator first.

```
vtest [" 3", OperatorQ[c0] == False]
vtest [" 4", CreateOperator[d0];
  OperatorQ[d0] == True]
```

Any expression containing an operator will be queried **True** by `OperatorQ[]`. This is so when the expression contains only an operator or when the expression contains any combination of scalars and operators. The only case in which `OperatorQ[]` returns **False** is if it is passed an expression containing only scalars.

```
vtest [" 5", OperatorQ[Times[Exp[d0], d0]] == True]
vtest [" 6", OperatorQ[Times[b0, d0]] == True]
vtest [" 7", OperatorQ[Times[a0, b0]] == False]
```

Operators created in a batch using

```
CreateOperator[matrix]
```

are assigned a *phylum* and an *order* which is determined by their position in the matrix.

```
CreateOperator[{{Ix, Iy, Iz},{Sx, Sy, Sz}}];
```

```
vtest["8", phylum[Ix] == 1]
vtest["9", order[Ix] == 1]
vtest["10", phylum[Iy] == 1]
vtest["11", order[Iy] == 2]
vtest["12", phylum[Sx] == 2]
vtest["13", order[Sx] == 1]
vtest["14", phylum[Ix] != phylum[Sx]]
vtest["15", order[Ix] == order[Sx]]
```

Another, more stringent test of the *phylum* and *order* system:

```
CreateOperator[{{Ix, Iy, Iz},{a, a†}}];

vtest["16", Not[Mult[a, a†] == aa†]]
vtest["17", phylum /@ {Ix, Iy, Iz} == {1, 1, 1}]
vtest["18", phylum /@ {a, a†} == {2, 2}]
vtest["19", order /@ {Ix, Iy, Iz} == {1, 2, 3}]
vtest["20", order /@ {a, a†} == {1, 2}]
```

Very that the operators created in a big list are in fact recognized as operators.

```
vtest["21", OperatorQ /@ {a0, Ix, a} == {False, True, True}]
```

5 Non-commutative multiplication

We next define our own non-commutative multiplication function, `Mult`. This is done in the `Mult.m` package. We provide a sorting function which rearranges operators being multiplied so that, if possible, operators of a higher *phylum* appear in front of operators of a lower *phylum*. During this rearrangement, non-commuting operators are not allowed to pass each other.

Listing 3: `unidyn/Mult.m`

First define bottom-out and empty cases:

```
Clear[Mult];
Mult[a0_] := a0
Mult[] := 1
```

`Mult` distributes over `Plus` and is associative:

```
Mult[a0____, b0_Plus, c0____] :=
  Plus @@ (Mult[a0, #, c0] & ) /@ List @@ b0
Mult[a0____, Mult[b0____, c0____], d0____] :=
  Mult[a0, b0, c0, d0]
```


Define rules for factoring out scalars:

```
Mult[a0____, b0_?ScalarQ, c0____] :=
  Times[b0, Mult[a0, c0]]
Mult[a0____, b0_?ScalarQ c0____, d0____] :=
  Times[b0, Mult[a0, c0, d0]]
```

The function `NCSort` will sort the operators in a list into canonical order, being careful not to allow non-commuting operators to “pass” each other. The list that you pass to `NCSort` must contain *only* operators. How this functions works is probably easiest seen with an example. Suppose the following operators were initialized as follows:

```
CreateOperator[{ {Ix, Iy, Iz}, {Sx, Sy, Sz} }]
```

Asked to sort the list of operators

$$a = \{S_x, I_y, I_x, S_y\},$$

we would report the sorted list

$$a_{\text{sorted}} = \{I_y, I_x, S_x, S_y\}.$$

The S operator, being of a higher phylum than an I operator, should be passed through the I operators. We do not pass I_x to the left of I_y because I_x and I_y do not commute.

In the code below, the statement `Map[phylum[#]&, a]` will create a list populated by the operators’ phyla, in this case $\{2, 1, 1, 2\}$. To create p we multiply by the number of operators in the list plus one, 5 in this example, and add to this $\{1, 2, 3, 4\}$. This gives $p = \{11, 7, 8, 14\}$. An operator’s ranking in this list depends on both its phylum and on it’s location in the original list. Sorting p will tell us how to order the operators in the original list a . The variable p_{new} keeps track of the locations of the elements of p in the *sorted* version of p ; in this example $p_{\text{new}} = \{3, 1, 2, 4\}$. We recognize p_{new} as the order in which the input operators should appear in a_{sorted} .

```
NCSort[a0_List] :=
Module[{n, p, anew, pnew},
  n = Length[a0];
  p = (n+1) Map[phylum[#]&, a0] + Table[i, {i, 1, n}];
  pnew = (Position[Sort[p], #] [[1, 1]]) & /@ p;
  anew = Table[0, {i, 1, n}];
  Do[anew[[pnew[[i]]]] = a0[[i]], {i, 1, n}];
  Return[anew]
```

The function `SortedMult` is the same as `Mult` except that it reorders the operators in the call list by applying `NCSort` before passing the result to `Mult`.

```
SortedMult[a0____] :=
  Mult[Sequence @@ NCSort[List[a0]]]
```

The function `MultSort` reorders all the operators in a `NonCommutativeMultiply` call.

```
MultSort[a0____] :=
  a0 /. Mult → SortedMult
```

Extensive unit testing shows that `Mult` distributes over addition, is associative, and pulls scalars out in front as desired. Testing is also carried out on the sorting function.

Listing 4: `unidyn/Mult-tests.m`

First define some operators and scalars.

```
CreateOperator[{{Ix, Iy, Iz},{Sx, Sy, Sz}}];
CreateScalar[{a0, b0, c0, d0}]
```

In the following test, the left hand side should resolve to `Mult[Ix, Iy]` while the right-hand side should resolve to `Times[Ix, Iy]`. These are *not* the same. If `Mult` does not properly recognize that `Ix` and `Iy` are operators and instead treats them as scalars, then the tests below will inadvertently pass.

```
vtest ["01", Not[Mult[Ix, Iy] === IxIy]]
```

Test that `Mult` distributes over addition and is associative. Note how `Mult` handles products of scalars and products of operators differently. Below we test for sameness (`===`) instead of equality (`==`). This is because if the two sides are *not* the same, then the equality test (using `==`) is undefined — neither true nor false — and the unit test does not fail as we wish.

```
vtest ["02a", Mult[a0, b0 + c0] === a0b0 + a0c0]
vtest ["02b", Mult[a0 + b0, c0] === a0c0 + b0c0]
```

Test that `Mult` distributes over addition and is associative. Note how `Mult` handles products of scalars and products of operators differently. Below we test for sameness (`===`) instead of equality (`==`). This is because if the two sides are *not* the same, then the equality test (using `==`) is undefined — neither true nor false — and the unit test does not fail as we wish.

```
vtest ["03", Mult[Iy, Ix] a0b0 === a0b0Mult[Iy, Ix]]

vtest ["04a", Mult[a0, b0 + c0] === a0b0 + a0c0]
vtest ["04b", Mult[a0 + b0, c0] === a0c0 + b0c0]

vtest ["04c", Mult[Ix, Iy + Iz] === Mult[Ix, Iy] + Mult[Ix, Iz]]
vtest ["04d", Mult[Ix + Iy, Iz] === Mult[Ix, Iz] + Mult[Iy, Iz]]
```

Here is an example where we apparently do *not* have to call `NCEExpand[]`.

```
vtest ["05a", Mult[Ix, Mult[Sx, Sy], Iz] === Mult[Ix, Sx, Sy, Iz]]
```

Test that scalars get factored out properly.

```
vtest ["06a", Mult[Ix, 2 I a0, Sx] === a02 I Mult[Ix, Sx]]
vtest ["06b", Mult[Ix, Mult[a0, Iy], Mult[b0, Sx]] === a0b0Mult[Ix, Iy, Sx]]
```

Test our sorting function:

```
vtest ["07a", NCSort[{Sx, Ix, Iy, Sz}] === {Ix, Iy, Sx, Sz}]
```

The function `MultSort[]` is used to order the operators in a standing `Mult[]` function.

```
vtest ["08a", SortedMult[Iy, Sx, Ix] === Mult[Iy, Ix, Sx]]
vtest ["08b", MultSort[Mult[Iy, Sx, Ix]] === Mult[Iy, Ix, Sx]]
```

If we now define the operators to have a different natural order, then the above tests fail. This confirms that the operators are being sorted according to our rules.

```
CreateOperator[{{Sx, Sy, Sz}, {Ix, Iy, Iz}}];
vtest ["08b", Not[SortedMult[Iy, Sx, Ix] === Mult[Iy, Ix, Sx]]]
vtest ["08d", Not[MultSort[Mult[Iy, Sx, Ix]] === Mult[Iy, Ix, Sx]]]
```

Check that `MultSort[]` works as expected when scalars are peppered into the list of operators being multiplied.

```
vtest ["09a", MultSort[Mult[a0 Iy, Sx, Ix]]
=== a0Mult[Sx, Iy, Ix]]

expr1 = Mult[Sx, a0Sz] + Mult[Ix, b0Sx];
expr2 = MultSort[Mult[expr1, Ix]];
expr3 = a0Mult[Sx, Sz, Ix] + b0Mult[Sx, Ix, Ix];

vtest ["10a", expr2 === expr3]
```

6 Commutator

Define a commutator function which factors out scalars and has rules for simplifying more complex commutators, like $[AB, C]$ and $[A, BC]$, involving three non-commuting operators.

Listing 5: unidyn/Comm.m

```
Comm[a0_?ScalarQ, b0_] := 0
Comm[a0_, b0_?ScalarQ] := 0
Comm[a0_, a0_] := 0
```

The commutator distributes over `Plus`:

```
Comm[a0___, b0_Plus] := Plus @@ (Comm[a0, #] &) /@ List @@ b0
Comm[a0_Plus, b0_] := Plus @@ (Comm[#, b0] &) /@ List @@ a0
```

Rules for factoring out scalars:

```
Comm[a0___ b0_?ScalarQ, c0_] := b0 Comm[a0, c0]
Comm[a0___, b0_?ScalarQ c0_] := b0 Comm[a0, c0]
```

Two rules for simplifying commutators involving products:

```
Comm[a0_Mult, C_] :=
  Module[{A,B},
    A = (List @@ a0)[[1]] ;
    B = Mult @@ Rest[List @@ a0] ;
    Mult[A, Comm[B, C]] + Mult[Comm[A, C], B]
  ]

Comm[A_, b0_Mult] :=
  Module[{B,C},
    B = (List @@ b0)[[1]] ;
    C = Mult @@ Rest[List @@ b0] ;
    Mult[Comm[A, B], C] + Mult[B, Comm[A, C]]
  ]
```

Finally, a commutator between different operators of a different *phyla* is zero:

```
Comm[a0_,b0_] := 0 /; phylum[a0] != phylum[b0];
```

Extensive testing shows that the commutator function factors out scalars correctly and correctly simplifies over a dozen commutation identities, including the Jacobi identity.

Listing 6: unidyn/Comm-tests.m

```
CreateScalar[{a0, b0, c0, d0}];
CreateOperator[{{I_x, I_y, I_z},{S_x, S_y, S_z}}];
```

The commutator is distributive. The commutator of a scalar with an operator is zero. The commutator of an operator with itself is zero. The commutator between two operators of different *phyla* is zero. The commutator of two operators of the same *phyla* is left unevaluated.

```
vtest["01a", Comm[I_x, I_y + I_z] === Comm[I_x, I_y] + Comm[I_x, I_z]]
vtest["01b", Comm[a0, I_x] === 0]
vtest["01c", Comm[I_x, I_x] === 0]
vtest["01d", Comm[I_x, S_y] === 0]
vtest["01e", Comm[S_x, S_y] === Comm[S_x, S_y]]
```

Scalars are properly factored out of commutators.

```
vtest["02a", Comm[a0 S_x, S_y] === a0 Comm[S_x, S_y]]
vtest["02b", Comm[S_x, b0 S_y] === b0 Comm[S_x, S_y]]
vtest["02c", Comm[a0 S_x, b0 S_y] === a0 b0 Comm[S_x, S_y]]
```

Test the $[AB, C]$ and $[A, BC]$ expansion rules, adding in scalars to both positions. In test 03b, we see that the S_x operator in the second place on the commutator can be pulled out front since it commutes with the operator in the first place of commutator.

```

vtest ["03a", Comm[a0 Mult[Sx, Sy], b0Sz]
=== a0b0(Mult[Sx, Comm[Sy, Sz]] + Mult[Comm[Sx, Sz], Sy])
vtest ["03b", Comm[a0 Sx, b0Mult[Sy, Sz]]
=== a0b0(Mult[Sy, Comm[Sx, Sz]] + Mult[Comm[Sx, Sy], Sz])
vtest ["03c", Comm[Sx, Mult[Sx, Sy]]
=== Mult[Sx, Comm[Sx, Sy]]

```

Test the Jacobi identity. For this identity to resolve to zero, we must tell *Mathematica* that $[A, B]$ equals $A*B-B*A$. First, check that this substitution does what we expect, then test the Jacobi identity.

```

vtest ["04a", (Comm[Ix, Iy] /. Comm[Sx_, Sy_] → Mult[Sx, Sy] - Mult[Sy, Sx])
=== Mult[Ix, Iy] - Mult[Iy, Ix]

vtest ["04b", ((Comm[Ix, Comm[Iy, Iz]] +
Comm[Iy, Comm[Iz, Ix]] + Comm[Iz, Comm[Ix, Iy]])
/. Comm[Sx_, Sy_] → Mult[Sx, Sy] - Mult[Sy, Sx])
=== 0]

```

Test the table of ten identities under “Advanced Identities” at the Wikipedia page for “Commutator”, <https://en.wikipedia.org/wiki/Commutator>. I find that the first equality number 9 in “Advanced Identities” does not actually evaluate to True.

```
CreateOperator[{{A, B, C, D, E$sym}}];
```

```

vtest ["05.01",
Comm[A, Mult[B, C]]
=== Mult[Comm[A, B], C]
+ Mult[B, Comm[A, C]]
]

vtest ["05.02",
Comm[A, Mult[B, C, D]]
=== Mult[Comm[A, B], C, D]
+ Mult[B, Comm[A, C], D]
+ Mult[B, C, Comm[A, D]]
]

vtest ["05.03",
Comm[A, Mult[B, C, D, E$sym]]
=== Mult[Comm[A, B], C, D, E$sym]
+ Mult[B, Comm[A, C], D, E$sym]
+ Mult[B, C, Comm[A, D], E$sym]
+ Mult[B, C, D, Comm[A, E$sym]]
]

vtest ["05.04",
Comm[Mult[A, B], C]
=== Mult[A, Comm[B, C]]
+ Mult[Comm[A, C], B]
]

```

```

vtest [" 05.05",
Comm[Mult[A, B, C], D]
=== Mult[A, B, Comm[C, D]]
+ Mult[A, Comm[B, D], C]
+ Mult[Comm[A, D], B, C]
]

vtest [" 05.06",
Comm[Mult[A, B, C, D], E$sym]
=== Mult[A, B, C, Comm[D, E$sym]]
+ Mult[A, B, Comm[C, E$sym], D]
+ Mult[A, Comm[B, E$sym], C, D]
+ Mult[Comm[A, E$sym], B, C, D]
]

vtest [" 05.07",
Comm[A, B+ C]
=== Comm[A, B]
+ Comm[A, C]
]

vtest [" 05.08",
Comm[A + B, C+ D]
=== Comm[A, C]
+ Comm[A, D]
+ Comm[B, C]
+ Comm[B, D]
]

vtest [" 05.09",
Comm[Mult[A, B], Mult[C, D]]
=== Mult[A, Comm[B, C], D]
+ Mult[A, C, Comm[B, D]]
+ Mult[Comm[A, C], D, B]
+ Mult[C, Comm[A, D], B]
]

vtest [" 05.10",
Module[{left, right },

left = Comm[Comm[A, C], Comm[B, D]]
//. Comm[A_?OperatorQ, B_?OperatorQ] → Mult[A,B] – Mult[B,A];

right = Comm[Comm[Comm[A, B], C], D]
+ Comm[Comm[Comm[B, C], D], A]
+ Comm[Comm[Comm[C, D], A], B]
+ Comm[Comm[Comm[D, A], B], C]
//. Comm[A_?OperatorQ, B_?OperatorQ] → Mult[A,B] – Mult[B,A];

left === right]
]

```

7 Spins

Declare angular momentum operators for one spin I_x , I_y , and I_z and specify the relevant commutation relations. To speed up computation, each operator's commutation relations are stored as an *upvalue* of the operator instead of as a *downvalue* of the commutator function. Additional product-operator simplification rules are defined if the spin is declared to have an angular momentum of $\ell = 1/2$.

Listing 7: unidyn/Spins.m

```
SpinSingle$CreateOperators[Ix_,Iy_,Iz_,L_:Null] :=
```

```
Module[{nonexistent},
```

Test if the operators exist; if they do not already exist, then create them. If an operator `Op` has been created already, then `CommutativeQ[Op]` will return `True`. Unless I_x , I_y , and I_z all already exist as operators, then create all three operators afresh. In the code below it is important that we call `Mult`CommutativeQ` and not just `CommutativeQ`.

```
nonexistent =
  Not[OperatorQ[Ix]] ||
  Not[OperatorQ[Iy]] ||
  Not[OperatorQ[Iz]];

If[nonexistent == True,
  Clear[Ix, Iy, Iz];
  CreateOperator[{{Ix, Iy, Iz}}];
  Message[SpinSingle$CreateOperators::create],
  Message[SpinSingle$CreateOperators::nocreate];];
```

The commutation relations are defined as *upvalues* of the various spin angular momentum operators. That is, the commutation relations are associated with the operators and not with the `Comm` function. The following commutation relations hold for any L .

```
Ix /: Comm[Ix,Iy] = I Iz;
Ix /: Comm[Ix,Iz] = -I Iy;
Iy /: Comm[Iy,Ix] = -I Iz;
Iy /: Comm[Iy,Iz] = I Ix;
Iz /: Comm[Iz,Iy] = -I Ix;
Iz /: Comm[Iz,Ix] = I Iy;
```

```
Message[SpinSingle$CreateOperators::comm]
```

```
Switch[L,
```

When the total angular momentum $L = 1/2$, additional rules are defined to simplify products of the angular momentum operators.

```
1/2,
```

```

Iz /: Mult[a___, Iz, Iz, b___] := Mult[a, b]/4;
Iy /: Mult[a___, Iy, Iy, b___] := Mult[a, b]/4;
Ix /: Mult[a___, Ix, Ix, b___] := Mult[a, b]/4;

Iz /: Mult[a___, Iz, Ix, b___] := I Mult[a, Iy, b]/2;
Iz /: Mult[a___, Iz, Iy, b___] := -I Mult[a, Ix, b]/2;

Iy /: Mult[a___, Iy, Ix, b___] := -I Mult[a, Iz, b]/2;
Iy /: Mult[a___, Iy, Iz, b___] := I Mult[a, Ix, b]/2;

Ix /: Mult[a___, Ix, Iz, b___] := -I Mult[a, Iy, b]/2;
Ix /: Mult[a___, Ix, Iy, b___] := I Mult[a, Iz, b]/2;

```

```
Message[SpinSingle$CreateOperators::simplify],
```

When the total angular momentum L is unspecified, no such simplification rules are defined.

```

Null,

Message[SpinSingle$CreateOperators::nosimplify]

];
Return[{Ix, Iy, Iz}]
]

```

Test the commutation relations for one spin and for a system of two spins, one of them $\ell = 1/2$ and the other with ℓ unspecified.

Listing 8: unidyn/Spins-tests.m

If there is one operator in the requested list of three spin operators that is not defined, then create all three spin operators afresh. Here we check that the test does what we want. If one of the operators is undefined, the test should come out false.

```

Clear[Ix, Iy, Iz];
CreateOperator[{{Ix, Iy}}];
tests = OperatorQ /@ {Ix, Iy, Iz};
vtest["00a", And @@ tests == False]

```

If, on the other hand, all three spin operators have been defined already, then the test should come out true.

```

Clear[Ix, Iy, Iz];
CreateOperator[{{Ix, Iy, Iz}}];
tests = OperatorQ /@ {Ix, Iy, Iz};
vtest["00b", And @@ tests == True]

```

We should also check the limiting case that **none** of the operators have been defined yet.

```

Clear[Ix, Iy, Iz];
tests = OperatorQ /@ {Ix, Iy, Iz};

```



```
vtest ["00c", And @@ Not /@ tests == True]
```

Create spin angular momentum operators with the total angular momentum unspecified. Test that the canonical angular momentum commutation relation holds true. With the total angular momentum unspecified, the product $I_x I_y$ cannot be simplified further.

```
Clear[Ix, Iy, Iz]
SpinSingle$CreateOperators[Ix, Iy, Iz];

vtest ["01a", Comm[Ix, Iy] === I Iz]
vtest ["01b", Not[Mult[Ix, Iy] === Mult[Iy, Ix]]]
vtest ["01c", Not[Mult[Ix, Iy] === I Iz/2]]
vtest ["01d", Not[Mult[Iz, Iz] === 1/4]]
```

Create two sets of operators, one set for I spins and one set for S spins. Assign the I -spin operators the properties of $I = 1/2$ angular momentum operators.

```
Clear[Ix, Iy, Iz, Sx, Sy, Sz];
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz, 1/2];
```

We have not defined the S operators to be spin operators yet. Nevertheless, the commutator of one S operator with another should be non-zero. On the other hand, the commutator of an S operator with an I operator should be zero.

```
vtest ["02a", Not[Comm[Sx, Sz] === 0]]
vtest ["02b", Comm[Iz, Sz] === 0]
```

The canonical commutation relations hold true for the I -spin operators. In addition, the product of two spin angular momentum operators can be further simplified.

```
vtest ["03a", Comm[Ix, Iy] === I Iz]
vtest ["03b", Mult[Ix, Iy] === I Iz/2]
vtest ["03c", Mult[Iz, Iz] === 1/4]
```

Test that the double commutator of I_x with the free-evolution Hamiltonian returns the expected result.

```
Clear[h, w, rho0, rho2];
CreateScalar[w];
h = w Iz;
rho0 = Ix;
rho2 = Comm[-I h, Comm[-I h, rho0]];
vtest ["04a", rho2 == - Mult[w, w, Ix]]
```

Try this test again with another spin operator on the backend.

```
Clear[Ix, Iy, Iz, Sx, Sy, Sz]
CreateOperator[{{Ix, Iy, Iz}, {Sx, Sy, Sz}}];
SpinSingle$CreateOperators[Ix, Iy, Iz];
SpinSingle$CreateOperators[Sx, Sy, Sz];
```

```

Clear[h, w, rho0, rho2];
CreateScalar[w];
h = w Mult[Iz, Sz];
rho0 = Ix;
rho2 = Comm[-I h, Comm[-I h, rho0]];
vtest["04b", rho2 == - Mult[w, w, Ix, Sz, Sz]]

```

8 Harmonic oscillator

Declare the annihilation (or lowering) operator a and the creation (or raising) operator a^\dagger for a harmonic oscillator, and specify the relevant commutation relations.

Listing 9: unidyn/Osc.m

The commutation relations are defined as *upvalues* of the lowering and raising operators. Here $aR = a$, the lowering operator, and $aL = a^\dagger$, the raising operator.

```

OscSingle$CreateOperators[a_, a†_] :=

Module[{nonexistent},

Test if the operators exist; if they do not already exist, then create them.

nonexistent =
  Not[OperatorQ[a]] ||
  Not[OperatorQ[a†]];

If [nonexistent == True,
  Clear[a, a†];
  CreateOperator[{{a, a†}}];
  Message[OscSingle$CreateOperators::create],
  Message[OscSingle$CreateOperators::nocreate];];

a /: Comm[a, a†] = 1;
a† /: Comm[a†, a] = -1;

Message[OscSingle$CreateOperators::comm]

Return[{a, a†}]
]

```

Check commutation relations among the harmonic-oscillator operators: a , a^\dagger , the number operator $N = a^\dagger a$, the unitless position operator $Q = (a^\dagger + a)/\sqrt{2}$, and the unitless momentum operator $P = i(a^\dagger - a)/\sqrt{2}$. Confirm that harmonic oscillator can exist “on top of” spin operators and are sorted correctly.

Listing 10: unidyn/Osc-tests.m

Create raising and lowering operators for a single harmonic oscillator. Check that the operators are indeed created. Check that they have the expected commutation relations.

```
Clear[a, a†, a0, Nop];

CreateScalar[{a0};
OscSingle$CreateOperators[a, a†];

vtest["01a", OperatorQ /@ {a, a†} == {True, True}]
vtest["01b", Comm[a, a†] === 1]
vtest["01c", Comm[a†, a] === -1]
```

Check that scalars are factored out of the commutations relations involving the harmonic oscillator raising and lower operators.

```
vtest["01d", Comm[a0 a, a†] === a0]
vtest["01e", Comm[a, a0 a†] === a0]
```

Test the commutation relations by defining the number operator, $Nop = N = a^\dagger a$, and checking the commutation relations $[N, a^\dagger] = a^\dagger$ and $[N, a] = -a$.

```
Nop = Mult[a†, a];

vtest["01d", Comm[Nop, a†] === a†]
vtest["01e", Comm[Nop, a] === -a]
```

```
Clear[a, a†, a0, Nop];
```

Check that we can define harmonic oscillator operators “on top of” an existing operator that commutes with the harmonic-oscillator operators.

```
Clear[a, a†, a, Q];

CreateScalar[{a}];
CreateOperator[{{Q},{a†, a}}]
OscSingle$CreateOperators[a, a†];

vtest["02a", Comm[a, a†] === 1]
vtest["02b", Comm[a†, a] === -1]
vtest["02c", Comm[a, Q] == 0]
vtest["02d", Comm[a, a] == 0]
vtest["02e", Comm[Q, a] === 0]
vtest["02f", MultSort[Mult[a, Q]] === Mult[Q, a]]
```

```
Clear[a, a†, a, Q];
```

Test the commutations for the position and momentum operators.

```
Clear[a, a†, Q, P];
```

```

OscSingle$CreateOperators[a, a†];
Q = (a† + a)/√2;
P = I (a† - a)/√2;

vtest["03a", Comm[Q, P] === I]
vtest["03b", Comm[P, Q] === -I]

Clear[a, a†, Q, P];

```

Another check that we can define harmonic oscillator operators “on top of” existing operators. In the following tests we prove that an operator like I_x (not defined as a spin operator, just an operator) commutes with one of the harmonic oscillator operators while the harmonic oscillator commutation relations are retained.

```

Clear[Ix, Iy, Iz, a, a†, Nop];
CreateOperator[{{Ix, Iy, Iz}, {a†, a}}]
OscSingle$CreateOperators[a, a†];
Nop = Mult[a†, a];

vtest["04a", Not[Comm[Ix, Iy] === 0]]
vtest["04b", Comm[Ix, a] == 0]
vtest["04c", Comm[Mult[Ix, Iy], a] == 0]
vtest["04d", Comm[a, a†] === 1]
vtest["04e", Comm[Nop, a†] === a†]
vtest["04f", Comm[Nop, a] === -a]

```

We defined the I operators to have higher precedence than the harmonic oscillator operators. Check that `MultSort` pulls the I operators out front as expected.

```

vtest["05", MultSort[Mult[a, Ix, a†, a, Iy]]
  === Mult[Ix, Iy, a, a†, a]]

Clear[Ix, Iy, Iz, a, a†, Nop];

```

9 Unitary Evolution

Implement the `Evolver` algorithm described above. The function can be passed an option to print out intermediate variables, which is helpful for debugging.

Listing 11: `unidyn/Evolve.m`

The evolution operator distribute over `Plus` and `Mult`.

```

Clear[Evolve];
Evolve[H_, t_, ρ_Plus] := Plus @@ (Evolve[H, t, #]&) /@ List @@ ρ

Evolve[H_, t_, ρ_Mult] := Mult @@ (Evolve[H, t, #]&) /@ List @@ ρ

```

Scalars in front of the density operator should be pulled out front.

```
Evolve[ $\mathcal{H}$ _, t_, Times[a_?ScalarQ,  $\rho$ _]] := a Evolve[ $\mathcal{H}$ , t,  $\rho$ ]
```

A test to see if all the terms in a sum commute with each other. It is important to include an `AllCommutingQ::usage` statement at the top of this package so that this function is available in the `General`` context of the notebook.

```
AllCommutingQ[ $\mathcal{H}$ _] := Module[{H$list, Comm$matrix},
  If [Head[ $\mathcal{H}$ ] === Plus,
    H$list = List @@  $\mathcal{H}$ ;
    Comm$matrix = Outer[Comm, H$list, H$list];
    Return[And @@ ((# === 0) & /@ Flatten[Comm$matrix])],
    Return[False]
  ]
]
```

If all the terms in the Hamiltonian commute, then we may distribute the `Evolve` operator over the terms in the Hamiltonian.

```
Evolve[ $\mathcal{H}$ _?AllCommutingQ, t_,  $\rho$ _] :=
  Mult @@ (Evolve[#, t,  $\rho$ ] &) /@ List @@  $\mathcal{H}$ 
```

A function to coerce *Mathematica* into writing simpler looking expressions, from <http://mathematica.stackexchange.com/questions/5403/how-to-get-fullsimplify-to-fully-simplify-my-expression-with-custom-complexity-f>

```
VisualComplexity:=(Count[ToBoxes[#],Except[" | "(" "]" , _String], Infinity ]&)
```

The `Evolver` function follows. By default, the function will not print out intermediate results during the computation.

```
Options[Evolver] = {quiet → True};

Evolver[ $\mathcal{H}$ _, t_,  $\rho_0$ _, opts : OptionsPattern[]] :=

Module[{k, a$vect, q, r, r$value, X, x4, x3, x2, x1, time, system, sol},

  Clear[ $\rho$ ];
   $\rho$ [0] =  $\rho_0$ ;
```

Evaluate the derivatives of the density operator by repeatedly applying the commutator. Simplify them as much as possible. Getting the simplification right is tricky. A special function has to be fed to `FullSimplify` to get it to return useful results.

```
Do[
   $\rho$ [k+1] = (-I Comm[ $\mathcal{H}$ ,  $\rho$ [k]] /. Mult → SortedMult)
    // FullSimplify [#, ComplexityFunction → VisualComplexity]&,
    {k,0,4}
];
```

Print out the vector of density-operator derivatives if asked.

```
If [OptionValue[quiet] == False, Print[" \[Rho] matrix = ",  $\rho$ [#]& /@
{0,1,2,3,4} // MatrixForm]];
```

Look for an entry in the $(\rho^{(2)}, \rho^{(1)}, \rho^{(0)})$ list that is proportional to $\rho^{(3)}$. Stop when you find it. Determining *proportional to* is tricky. Here we use the ability of the `NCAIgebra` package to compute a symbolic inverse of an operator. When $(\rho^{(n)})^{-1} * \rho^{(3)}$ is a scalar, then we have found a match. We are implicitly assuming that the entries $(\rho^{(2)}, \rho^{(1)}, \rho^{(0)})$ have an inverse. This will be true of then entries involve Hermitian operators. If the entries involve *non-Hermitian operators*, however, like I_+ or I_- then the entries might not have a proper inverse. We do not, at present, test whether the operators in the list are Hermitian or not.

```
a$vect={0,0,0,0};
r = Null;
r = Catch[
  Do[
    q = Mult[
      Divide[ $\rho$ [3], Mult[ $\rho$ [k]]]
    ] // FullSimplify [# , ComplexityFunction  $\rightarrow$  VisualComplexity]& ;

    If [ScalarQ[q]==True, Throw[{3-k, q}]],
    {k,0,2}
  ]
];
```

If we have not found a match by now, then throw up our hands and exit the function. Spit out the full $(\rho^{(3)}, \rho^{(2)}, \rho^{(1)}, \rho^{(0)})$ vector, in case the user can spot the *proportional to* condition.

```
If [r=== Null,
  Message[Evolver::unsolvable];
  Return[ $\rho$ [#]& /@ {0,1,2,3,4}],
  r$value=r
];
```

Set up the coupling matrix Ω based on the matching condition.

```
a$vect[[r$value [[1]]]] = r$value [[2]];
A = {a$vect,{1, 0 ,0, 0},{0, 1, 0, 0}, {0, 0 ,1, 0}};
```

If asked, spit out the coupling matrix for inspection.

```
If [OptionValue[quiet] == False, Print[" \[CapitalOmega] = ", A // MatrixForm]];
```

Set up the four coupled equations and solve them. Respect *Mathematica* version 8 standards here and feed `DSolve[]` a list where each element of the list is an equation. First set up the equations.

```
X[time_] = {x4[time], x3[time], x2[time], x1[time]};
lhs$list = D[X[time],time];
rhs$list = A . X[time];
eqns = ( lhs$list [[#]] == rhs$list [[#]]& /@ {1,2,3,4};
```

```

system = {eqns,
  x4[0]== ρ[3], x3[0]== ρ[2], x2[0]== ρ[1], x1[0]== ρ[0]];

If [OptionValue[quiet] == False, Print[" system of equations = ", system // MatrixForm]];

```

Now solve them and print out the solution with the private variables replaced by public ones.

```

sol = DSolve[system,{x1,x2,x3,x4},time];

If [OptionValue[quiet] == False, Print[" 1st solution = ", sol [[1]][[1]]  ]];
If [OptionValue[quiet] == False, Print[" 1st solution w/ substitution = ", x1[time]
/. sol [[1]][[1]] /. time → t]];

```

Return only the first element of the solution, $\lambda_1(t)$.

```

Return[FullSimplify[x1[time] /. sol [[1]] /. time → t]];
];

```

Unit testing includes verifying that `Evolver` distributes over sums and factors our scalars correctly. Unit testing also verifies that *Mathematica*'s differential equation solver is behaving as expected. We are not worried about bugs in the solver. We are instead concerned about changes in function syntax. The solver function's syntax changed between *Mathematica* version 8 and 10 and could change again. We added an equation-solver unit test to flag a future syntax change.

We test the following cases of unitary evolution in the rotating frame:

1. $e^{-i\omega t I_z} I_x e^{+i\omega t I_z}$
free evolution of transverse magnetization in a magnetic field
2. $e^{-i\omega t I_x} I_z e^{+i\omega t I_x}$
on-resonance nutation of longitudinal (i.e., thermal-equilibrium) magnetization in the rotating frame
3. $e^{-i\omega t I_z} I_+ e^{+i\omega t I_z}$
free evolution of complex transverse magnetization in a magnetic field
4. $e^{-idt I_z S_z} I_x e^{+idt I_z S_z}$
evolution of transverse magnetization under a scalar coupling
5. $e^{-i(\Delta I_z + \omega I_x)t} I_z e^{+i(\Delta I_z + \omega I_x)t}$
off-resonance nutation of longitudinal magnetization
6. $e^{-\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)} a e^{+\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)}$
evolution of the lowering (and raising) operator under the harmonic oscillator Hamiltonian

7. $e^{-\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)} Q e^{+\omega t \frac{1}{2}(a^\dagger a + a a^\dagger)}$
evolution of the position (and momentum) operator under the harmonic oscillator Hamiltonian
8. $e^{-i\delta q P} Q e^{+i\delta q P}$
a position kick
9. $e^{-i\delta q Q} P e^{+i\delta q Q}$
a momentum kick

The off-resonance spin nutation, case 5, is an especially demanding test because it generates terms involving I_x , I_y , and I_z operators from solutions to a third-order differential equation.

Listing 12: unidyn/Evolve-tests.m

Global debugging flag to force `Evolver` to be noisy instead of quiet. Set to `False` for noisy output and to `True` for quiet output.

```
quiet$Query = True;
```

Show that the `Evolve` operator is distributive over both sums and non-commutative products of operators. Show that scalars are pulled out front, even if buried in a complicated product of operators and scalars. For this test, we'll make three sets of two commuting operators representing, for example, three independent harmonic oscillators.

```
Clear[H, t, H1, H2, H3, Q, R, S, U, V, W, q, r, s, u, v, w];
```

```
CreateOperator[{{Q,R},{S,U},{V,W}}]
CreateScalar[{q,r,s,u,v,w}]
```

```
vtest["01a > distribute addition",
  Evolve[H, t, Q + R + S]
  === Evolve[H, t, Q] + Evolve[H, t, R] + Evolve[H, t, S]]
vtest["01b > distribute multiplication ",
  Evolve[H, t, Mult[Q, R, S]]
  === Mult[Evolve[H, t, Q], Evolve[H, t, R], Evolve[H, t, S]]]
vtest["01c > distribute complicated expression",
  Evolve[H, t, Mult[(Q q), (r R), (s S)] + u U]
  === u Evolve[H, t, U] + q r s Mult[Evolve[H, t, Q], Evolve[H, t, R], Evolve[H, t, S]]]
```

Make up some Hamiltonians and see if they pass the all-terms-commuting test. The first test is particularly important. In test 03a and 03b below, nothing happens; the Hamiltonian is either so simple that it can be broken into pieces, 03a, or contains terms which do not commute, 03b. In test 03c we have a Hamiltonian whose three terms commute, and in this case the `Evolve` operator can be expanded.

```
H0 = Q;
H1 = q Mult[Q, R] + s Mult[S, U] + Mult[U, S] + Mult[V, V, W];
H2 = q Q + s S + v Mult[Q, S];
```



```

vtest["02a > commuting test 1", AllCommutingQ[H0] === False]
vtest["02b > commuting test 2", AllCommutingQ[H1] === False]
vtest["02c > commuting test 3", AllCommutingQ[H2] === True]

vtest["03a > Evolve expand test 1", Evolve[H0, t, Q] === Evolve[Q, t, Q]]
vtest["03b > Evolve expand test 2", Evolve[H1, t, Q]
=== Evolve[q Mult[Q,R] + s Mult[S,U] + Mult[U, S] + Mult[V, V, W], t, Q]]
vtest["03c > Evolve expand test 3", Evolve[H2, t, Q]
=== Mult[Evolve[q Q, t, Q], Evolve[s S, t, Q], Evolve[v Mult[Q,S], t, Q]]]

Clear[H0, H1, H2]
Clear[Q, R, S, U, V, W, q, r, s, u, v, w]

```

To test the unitary evolution operator on spins, let us set up a model two-spin system. This system is comprised of an $L = 1/2$ I spin and an unspecified- L S spin.

```

Clear[ $I_x, I_y, I_z, S_x, S_y, S_z, \omega, d_0, \Delta, \rho, t$ ]
Clear[A, r, lhs$list, rhs$list, time, eqns, system,  $\rho_{\text{calc}}, \rho_{\text{known}}, X$ ]

CreateScalar[ $\omega, d_0, \Delta$ ];
CreateOperator[{{ $I_x, I_y, I_z$ },{ $S_x, S_y, S_z$ }}]

SpinSingle$CreateOperators[ $I_x, I_y, I_z, L=1/2$ ];
SpinSingle$CreateOperators[ $S_x, S_y, S_z, L=1/2$ ];

```

Test the differential equation solver first. In *Mathematica* version 10 gives more leeway in how the equations are set up – you can set one list equal to another, for example. In *Mathematica* version 8, in contrast the syntax is not so forgiving. Let us mock-up an equation by hand and feed it to the solver.

```

A = {{0, - $\Delta^2, 0, 0$ }, {1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}};
r = { $I_x, I_y \Delta, -I_x \Delta^2, -I_y \Delta^3, I_x \Delta^4$ };
( $\rho[\#-1] = r[[\#]]$ ) & /@ {1, 2, 3, 4, 5};

X[time_] = {x4[time], x3[time], x2[time], x1[time]};
lhs$list = D[X[time], time];
rhs$list = A . X[time];
eqns = (lhs$list[[#]] == rhs$list[[#]]) & /@ {1, 2, 3, 4};

system = {eqns,
  x4[0] ==  $\rho[3]$ , x3[0] ==  $\rho[2]$ , x2[0] ==  $\rho[1]$ , x1[0] ==  $\rho[0]$ };
sol = DSolve[system, {x1, x2, x3, x4}, time];

 $\rho_{\text{calc}} = (x1[time] /. sol[[1]] /. time \rightarrow t) // \text{Expand} // \text{ExpToTrig} // \text{FullSimplify}$ ;
 $\rho_{\text{known}} = I_x \mathbf{Cos}[t \Delta] + I_y \mathbf{Sin}[t \Delta]$ ;

vtest["04a > DSolve test",  $\rho_{\text{calc}} == \rho_{\text{known}}$ ]

```

Free evolution of I_x :

```

vtest["05a > free evolution of  $I_x$ ",
  FullSimplify[ExpToTrig[Expand[

```

```

Evolver[ $\omega$   $I_z$ ,  $t$ ,  $I_x$ , quiet  $\rightarrow$  quiet$query]]]]
===  $I_x \mathbf{Cos}[\omega t] + I_y \mathbf{Sin}[\omega t]$ 

```

On-resonance nutation of I_z :

```

vtest["05b > on-resonance nutation of  $I_z$ ",
FullSimplify[ExpToTrig[Expand[
Evolver[ $\omega$   $I_x$ ,  $t$ ,  $I_z$ , quiet  $\rightarrow$  quiet$query]]]]
===  $I_z \mathbf{Cos}[\omega t] - I_y \mathbf{Sin}[\omega t]$ 

```

Free evolution of I_+ :

```

vtest["05c > free evolution of  $I_+$ ",
FullSimplify[Evolver[ $\omega$   $I_z$ ,  $t$ ,  $I_x$ , quiet  $\rightarrow$  quiet$query]
+  $I$  Evolver[ $\omega$   $I_z$ ,  $t$ ,  $I_y$ , quiet  $\rightarrow$  quiet$query]]
===  $\mathbf{Exp}[-i \omega t](I_x + i I_y)$ 

```

Evolution under a scalar coupling:

```

vtest["05d > scalar-coupling evolution of  $I_x$ ",
Evolver[ $d_0$  Mult[ $I_z$ ,  $S_z$ ],  $t$ ,  $I_x$ , quiet  $\rightarrow$  quiet$query]
===  $I_x \mathbf{Cos}[d_0 t/2] + 2 \text{Mult}[I_y, S_z] \mathbf{Sin}[d_0 t/2]$ 

```

Off-resonance nutation of I_z . It is important to carefully set the assumptions used by Simplify.

```

$Assumptions = {Element[ $\Delta$ , Reals],  $\Delta > 0$ , Element[ $\omega$ , Reals],  $\omega >= 0$ };

```

```

 $c_1 = (\Delta^2)/(\Delta^2 + \omega^2)$ ;
 $c_2 = (\Delta \omega)/(\Delta^2 + \omega^2)$ ;
 $c_3 = (\omega^2)/(\Delta^2 + \omega^2)$ ;
 $c_4 = \omega/\text{Sqrt}[\Delta^2 + \omega^2]$ ;
 $\omega_{\text{eff}} = \text{Sqrt}[\Delta^2 + \omega^2]$ ;

```

```

 $\rho_{\text{known}} = \text{Collect}[$ 
 $c_1 I_z + c_2 I_x + (c_3 I_z - c_2 I_x) \mathbf{Cos}[\omega_{\text{eff}} t] - c_4 I_y \mathbf{Sin}[\omega_{\text{eff}} t]$  //
Expand, { $I_x$ ,  $I_y$ ,  $I_z$ });

```

```

 $\rho_{\text{calc}} = \text{Collect}[$ 
Evolver[ $\Delta I_z + \omega I_x$ ,  $t$ ,  $I_z$ , quiet  $\rightarrow$  quiet$query]
// FullSimplify, { $I_x$ ,  $I_y$ ,  $I_z$ }, Expand];

```

```

vtest["05e > Off-resonance nutation of  $I_z$ ",  $\rho_{\text{calc}} === \rho_{\text{known}}$ ]

```

Clean up:

```

Clear[ $I_x$ ,  $I_y$ ,  $I_z$ ,  $S_x$ ,  $S_y$ ,  $S_z$ ,  $\omega$ ,  $d_0$ ,  $\Delta$ ,  $\rho$ ,  $t$ ]
Clear[A, r, lhs$list, rhs$list, time, eqns, system,  $\rho_{\text{calc}}$ ,  $\rho_{\text{known}}$ , X]

```

Harmonic oscillator evolution. First, create the harmonic oscillator Hamiltonian in symmetric form. Evolve the lowering operator and confirm that it picks up the expected phase factor. Evolve the raising operator and confirm that it picks up the expected (conjugate) phase factor.

```

Clear[a, a†, ω, Q, P,  $\mathcal{H}$ , Q, P, δq, δp];
CreateScalar[ω, delta $\times$ sym, δp];
OscSingle$CreateOperators[a, a†];

 $\mathcal{H}$  = ω (Mult[a, a†] + Mult[a†, a])/2;

vtest["06a1 > free evolution of lowering operator",
  Simplify[TrigToExp[Expand[Evolver[ $\mathcal{H}$ , t, a, quiet → quiet$query]]]]
  === a Exp[I ω t]]

vtest["06a2 > free evolution of raising operator",
  Simplify[TrigToExp[Expand[Evolver[ $\mathcal{H}$ , t, a†, quiet → quiet$query]]]]
  === a† Exp[-I ω t]]

```

Evolve the position operator. To do this, write the position operator in terms of the raising and lowering operators, calculate the evolution, and rewrite the answer in terms of the position and momentum operator. For the re-write step, create a *new* set of non-commuting operators, to avoid chasing our tail.

```

CreateOperator[{{Q,P}}];
{Q, P} = {(a† + a)/√2, I (a† - a)/√2};
QP$rules = {a† → (Q - I P)/√2, a → (Q + I P)/√2};

```

Write Q and P in terms of raising and lower operators, write the raising and lowering operators in terms of position and momentum, and you should get the original Q and P back again.

```

vtest["06b > test Q definition ", Simplify[Q /. QP$rules] === Q]
vtest["06c > test P definition ", Simplify[P /. QP$rules] === P]

```

Now we are ready to evolve position and momentum.

```

vtest["06d > free evolution of Q",
  Simplify[ExpToTrig[Expand[Evolver[ $\mathcal{H}$ , t, Q, quiet → quiet$query] /. QP$rules]]]
  == Q Cos[ω t] - P Sin[ω t]]

vtest["06e > free evolution of P",
  Simplify[ExpToTrig[Expand[Evolver[ $\mathcal{H}$ , t, P, quiet → quiet$query] /. QP$rules]]]
  == P Cos[ω t] + Q Sin[ω t]]

```

Check that the operator $e^{-i\delta q P}$ delivers a position kick and that the operator $e^{-i\delta p X}$ delivers a momentum kick. These are examples of evolution where the commuting series terminates.

```

vtest["06f > position kick",
  Simplify[Evolver[δq P, t, Q] /. QP$rules ~Join~ {t → 1}]
  == Q - δq]

```

```
vtest["06g > momentum kick",
  Simplify[Evolver[ $\delta p$  Q, t, P] /. QP$rules ~Join~ {t  $\rightarrow$  1}]
  == P +  $\delta p$ ]
```

Clean up:

```
Clear[a,  $a^\dagger$ ,  $\omega$ , Q, P,  $\mathcal{H}$ , Q, P,  $\delta q$ ,  $\delta p$ ];
```

References

- [1] C. P. Slichter, *Principles of Magnetic Resonance, 3rd Edition*, 1990.
- [2] J. W. Shriver, NMR product-operator calculations in *Mathematica*, *J. Magn. Res.*, **1991**, 94, 612–616, URL [https://doi.org/10.1016/0022-2364\(91\)90150-R](https://doi.org/10.1016/0022-2364(91)90150-R).
- [3] R. P. F. Kanthers, B. W. Char, and A. W. Addison, A computer-algebra application for the description of NMR experiments using the product-operator formalism, *J. Magn. Reson. Ser. A*, **1993**, 101, 23–29, URL <https://doi.org/10.1006/jmra.1993.1003>.
- [4] P. Güntert, N. Schaefer, G. Otting, and K. Wuthrich, POMA: A complete *Mathematica* implementation of the NMR product-operator formalism, *J. Magn. Reson. Ser. A*, **1993**, 101, 103–105, URL <https://doi.org/10.1006/jmra.1993.1016>.
- [5] D. Isbister, M. S. Krishnan, and B. Sanctuary, Use of computer algebra for the study of quadrupole spin systems, *Molec. Phys.*, **1995**, 86, 1517–1535, URL <https://doi.org/10.1080/00268979500102891>.
- [6] I. Rodriguez and J. Ruiz-Cabello, Density matrix calculations in *Mathematica*, *Concepts Magn. Res.*, **2001**, 13, 143–147, URL [https://doi.org/10.1002/1099-0534\(2001\)13:2<143::AID-CMR1003>3.0.CO;2-4](https://doi.org/10.1002/1099-0534(2001)13:2<143::AID-CMR1003>3.0.CO;2-4).
- [7] P. Güntert, Symbolic NMR product operator calculations, *Int. J. Quantum Chem.*, **2006**, 106, 344–350, URL <https://doi.org/10.1002/qua.20754>.
- [8] C. K. Anand, A. D. Bain, and Z. Nie, Simulation of steady-state NMR of coupled systems using Liouville space and computer algebra methods, *J. Magn. Res.*, **2007**, 189, 200–208, URL <https://doi.org/10.1016/j.jmr.2007.09.012>.
- [9] I. Kuprov, N. Wagner-Rundell, and P. J. Hore, Bloch-Redfield-Wangsness theory engine implementation using symbolic processing software, *J. Magn. Res.*, **2007**, 184, 196–206, URL <https://doi.org/10.1016/j.jmr.2006.09.023>.
- [10] X. Filip and C. Filip, SD-CAS: Spin dynamics by computer algebra system, *J. Magn. Res.*, **2010**, 207, 95–113, URL <https://doi.org/10.1016/j.jmr.2010.08.014>.
- [11] T. Loke and J. B. Wang, An efficient quantum circuit analyser on qubits and qudits, *Comput. Phys. Commun.*, **2011**, 182, 2285–2294, URL <https://doi.org/10.1016/j.cpc.2011.06.001>.
- [12] Y. G. Chen and J. B. Wang, *Qcompiler*: Quantum compilation with the CSD method, *Comput. Phys. Commun.*, **2013**, 184, 853–865, URL <https://doi.org/10.1016/j.cpc.2012.10.019>.

- [13] T. Loke and J. B. Wang, *CUGatesDensity* — quantum circuit analyser extended to density matrices, *Comput. Phys. Commun.*, **2013**, 184, 2834–2839, URL <https://doi.org/10.1016/j.cpc.2013.07.007>.
- [14] R. Žitko, SNEG — Mathematica package for symbolic calculations with second-quantization-operator expressions, *Comput. Phys. Commun.*, **2011**, 182, 2259–2264, URL <https://doi.org/10.1016/j.cpc.2011.05.013>.
- [15] A. N. F. Aleixo and A. B. Balantekin, Exponential operators and the algebraic description of quantum confined systems, *J. Math. Phys.*, **2011**, 52, 082106, URL <https://doi.org/10.1063/1.3625627>.
- [16] V. N. Beskrovnyi, Applying Mathematica to the analytical solution of the nonlinear Heisenberg operator equations, *Comput. Phys. Commun.*, **1998**, 111, 76–86, URL [https://doi.org/10.1016/S0010-4655\(98\)00029-0](https://doi.org/10.1016/S0010-4655(98)00029-0).
- [17] N.-A. Nguyen and T. T. Nguyen-Dang, Symbolic calculations of unitary transformations in quantum dynamics, *Comput. Phys. Commun.*, **1998**, 115, 183–199, URL [https://doi.org/10.1016/S0010-4655\(98\)00129-5](https://doi.org/10.1016/S0010-4655(98)00129-5).
- [18] T. S. Untidt and N. C. Nielsen, Closed solution to the Baker-Campbell-Hausdorff problem: Exact effective Hamiltonian theory for analysis of nuclear-magnetic-resonance experiments, *Phys. Rev. E*, **2002**, 65, 021108, URL <https://doi.org/10.1103/PhysRevE.65.021108>.
- [19] D. Siminovitch, T. Untidt, and N. C. Nielsen, Exact effective Hamiltonian theory. II. Polynomial expansion of matrix functions and entangled unitary exponential operators, *J. Chem. Phys.*, **2004**, 120, 51–66, URL <https://doi.org/10.1063/1.1628216>.
- [20] M. Weyrauch and D. Scholz, Computing the Baker–Campbell–Hausdorff series and the Zassenhaus product, *Comput. Phys. Commun.*, **2009**, 180, 1558–1565, URL <https://doi.org/10.1016/j.cpc.2009.04.007>.
- [21] F. Casas, A. Murua, and M. Nadinic, Efficient computation of the Zassenhaus formula, *Comput. Phys. Commun.*, **2012**, 183, 2386–2391, URL <https://doi.org/10.1016/j.cpc.2012.06.006>.
- [22] A. Jerschow, MathNMR: Spin and spatial tensor manipulations in Mathematica, *J. Magn. Res.*, **2005**, 176, 7–14, URL <https://doi.org/10.1016/j.jmr.2005.05.005>.
- [23] C. Bengs and M. H. Levitt, SpinDynamica: Symbolic and numerical magnetic resonance in a mathematica environment, *Mag. Res. Chem.*, **2017**, 56, 374–414, URL <https://doi.org/10.1002/mrc.4642>.
- [24] P.-O. Löwdin, Quantum theory of many-particle systems. III. Extension of the Hartree-Fock scheme to include degenerate systems and correlation effects, *Phys. Rev.*, **1955**, 97, 1509–1520, URL <https://doi.org/10.1103/PhysRev.97.1509>.

- [25] J. W. Helton, M. de Oliveira, M. Stankus, and R. L. Miller, NCAgebra — Version 5.0.6, 2015.